

# Elaine: Elaborations of Higher-Order Effects as First-Class Language Feature



Terts Diepraam  
Master's Thesis

---

# Elaine: Elaborations of Higher-Order Effects as First-Class Language Feature

---

THESIS

submitted in partial fulfillment of the  
requirements for the degree of

MASTER OF SCIENCE

in

COMPUTER SCIENCE

by

Terts Diepraam  
born in Amsterdam, the Netherlands



Programming Languages Group  
Department of Software Technology  
Faculty EEMCS, Delft University of Technology  
Delft, the Netherlands  
[www.ewi.tudelft.nl](http://www.ewi.tudelft.nl)

© 2023 Terts Diepraam.

Cover picture: Rubin's Vase.

---

# Elaine: Elaborations of Higher-Order Effects as First-Class Language Feature

---

Author: Terts Diepraam  
Student id: 5652235

## Abstract

*Algebraic effects* and *handlers* have become a popular abstraction for effectful computation, with implementations even in mainstream programming languages, such as OCaml. The operations of an algebraic effect define the syntax of the effect, while handlers define the semantics. This provides modularity, because we can choose which handler to apply to a computation. However, we cannot write handlers for many *higher-order operations*; operations that take effectful computations as parameters. Such higher-order operations can therefore not enjoy this modularity. *Hefty algebras* provide an additional layer of abstraction in the form of *elaborations* to make implementations of higher-order operations modular as well. Several languages, such as Koka, natively support algebraic effects and handlers. However, until now, no languages have been created with native support for higher-order effects. In this thesis, we introduce Elaine, a language featuring both handlers for algebraic effects and elaborations for higher-order effects. Additionally, we introduce *implicit elaboration resolution*; a type-directed procedure which infers the appropriate elaborations from context. We conjecture that hefty algebras are the semantics for Elaine. We provide a specification for Elaine, including its syntax definition, typing judgments and reduction semantics. This specification is implemented in a publicly available prototype which can type check and evaluate the set of example programs included with this thesis.

Thesis Committee:

Chair: Prof. dr. M. Spaan, Faculty EEMCS, TU Delft  
Committee Member: Dr. C. Bach Poulsen, Faculty EEMCS, TU Delft  
University Supervisor: MSc. C. van der Rest, Faculty EEMCS, TU Delft



---

# Preface

There is a curse surrounding algebraic effects: once you know about them, you want to use them everywhere. Since I started writing this thesis, I wished countless times that some language I was working with had support algebraic effects or elaborations. So, I think it is incredibly exciting that mainstream language are starting to pick up on the research done on algebraic effects and are slowly adopting this feature. I hope that this thesis can contribute ever so slightly to this adoption.

In any case, tread carefully reading this thesis, for you may be cursed.

Terts Diepraam  
Delft, the Netherlands  
September 23, 2023

**About the cover illustration** The cover image depicts Rubin’s Vase in the shape of the Holy Grail. Rubin’s Vase is a shape that is bi-stable, meaning that it can be viewed in two ways. One might see either the two faces or the vase. This represents how effectful computations in Elaine can be given different interpretations with different handlers and elaborations. This version of Rubin’s Vase is meant to resemble the Holy Grail. After having chosen the name Elaine for the language developed for this thesis, we learned that several characters in Arthurian legend are called Elaine. One of these characters is Elaine of Corbenic, who is also called the “Grail Bearer” or the “Grail Maiden”.



---

# Contents

<b>Preface</b>	<b>iii</b>
<b>Contents</b>	<b>v</b>
<b>1 Introduction</b>	<b>1</b>
<b>2 Algebraic Effects</b>	<b>7</b>
2.1 Monads . . . . .	7
2.2 Composing Effects with the Free Monad . . . . .	11
2.3 Algebraic Effects . . . . .	14
2.4 Building a Language with Algebraic Effects . . . . .	15
<b>3 Higher-Order Effects</b>	<b>19</b>
3.1 Computation Parameters . . . . .	19
3.2 Breaking Algebraicity . . . . .	20
3.3 The Modularity Problem . . . . .	20
3.4 Hefty Algebras . . . . .	21
<b>4 A Tour of Elaine</b>	<b>27</b>
4.1 Overview . . . . .	27
4.2 Basics . . . . .	27
4.3 Types . . . . .	30
4.4 Algebraic Data Types . . . . .	30
4.5 Recursion & Loops . . . . .	31
4.6 Modules . . . . .	32
4.7 Algebraic Effects . . . . .	33
4.8 Functions Generic over Effects . . . . .	36
4.9 Higher-Order Effects . . . . .	38
<b>5 Implicit Elaboration Resolution</b>	<b>43</b>
<b>6 Elaine Specification</b>	<b>45</b>
6.1 Syntax Definition . . . . .	45
6.2 Effect Rows . . . . .	45
6.3 Type System . . . . .	47
6.4 Desugaring . . . . .	50
6.5 Semantics . . . . .	50
6.6 Standard Library . . . . .	51



<b>7</b>	<b>Related Work</b>	<b>55</b>
7.1	Monad Transformers . . . . .	55
7.2	Other Solutions to the Modularity Problem . . . . .	56
7.3	Languages with Native Support for Algebraic Effects . . . . .	57
7.4	Effects as Free Monads . . . . .	58
<b>8</b>	<b>Conclusion</b>	<b>59</b>
8.1	Future Work . . . . .	59
	<b>Bibliography</b>	<b>61</b>
<b>A</b>	<b>Elaine Example Programs</b>	<b>65</b>
A.1	A naive SAT solver . . . . .	65
A.2	Reader Effect . . . . .	66
A.3	Writer Effect . . . . .	67
A.4	Structured Logging . . . . .	68
A.5	Parser Combinators . . . . .	69

# Chapter 1

---

## Introduction

In many programming languages, computations are allowed to have *effects*. This means that they can perform operations besides producing output, and interact with their environment. A computation might, for instance, read or modify a global variable, write to a file, throw an exception, or even exit the program.

Historically, programming languages have supported effects in different ways. Some programming languages opt to give the programmer virtually unrestricted access to effectful operations. For instance, any part of a C program can interact with memory, the filesystem, or the network. The program can even yield control to any location in program with the `goto` keyword, which has famously been criticized by Dijkstra (1968), who argued that `goto` breaks the structure of the code and therefore makes the program harder to understand. The same reasoning extends to other effects: the more effects a function is allowed to exhibit, the harder it becomes to reason about.

The “anything goes” approach to effects therefore puts a large burden of ensuring correct behaviour of a program on the programmer. If the language cannot provide any guarantees about what (a part of) a program can do, the programmer has to check instead. For instance, if a function somewhere in the code sets global variable to some incorrect value. This can then cause seemingly unrelated parts of the program to behave incorrectly. The programmer tasked with debugging this issue then has to examine the program as a whole to find where this modification takes place. In languages where this is possible, effectful operations therefore limit our ability to split the code into chunks to be examined separately.

A solution is to treat effects in a more structured manner. For example, instead of allowing `goto` for error handling, a language might provide exceptions. In a language like Java, checked exceptions are part of the type system, so that the type checker can verify that all exceptions are handled. However, with this approach, any effect must be backed by the language. That is, the language needs to have a dedicated feature for every effect that should be supported, and new effects cannot be created without adding a new feature to the language. This means that the support for various effects is limited to what the language designers have decided to add.

In contrast, languages adhering to the functional programming paradigm disallow effectful operations altogether.<sup>1</sup> Here, all functions are *pure*, meaning that they always return identical outputs for identical inputs and do not interact with the environment. By requiring that all functions are pure, a type signature of a function becomes almost a full specification of what the function can do.

However, effectful operations are often still desired. Consider the following program written in Koka, a functional language where functions need to be pure. In this program,

---

<sup>1</sup>Usually there are some escape hatches to this rule, such as Haskell’s `trace` function, which is built-in and effectful, but only supposed to be used for debugging.

there is a set of users that are considered administrators. The `all_admins` function checks whether all user ID's in a list are administrators.

```
1 val admins = [0,1,2]
2
3 fun is_admin(user_id: int): bool
4   admins.any(fn(x) x == user_id)
5
6 fun all_admins(l: list<int>): bool
7   l.map(is_admin).foldl(True, (&&))
8
9 val result = all_admins([0,1,2,3])
```

Now, we would like the program to tell us which users are not admins. In an imperative language, we could just add a `print` call in `is_admin` to log any user that was not an admin and call it a day. But in a functional language, that is not possible. Instead, each message we want to log needs to be returned by the `is_admin`. These messages then need to be concatenated to build up the string that should be printed.

```
1 fun is_admin(user_id: int): (bool, string)
2   if admins.any(fn(x) x == user_id)
3   then (True, "")
4   else (False, "Denied " ++ show(user_id) ++ "\n")
5
6 fun all_admins(list: list<int>): (bool, string)
7   match list
8   Nil() -> (True, "")
9   Cons(x, xs) ->
10     val (y, s) = is_admin(x)
11     val (ys, s') = all_admins(xs)
12     (y && ys, s ++ s')
13
14 val (result, log) = all_admins([1,2,3,4])
```

Adding some logging has made the program much more complicated than it was originally. For larger programs, one might imagine that programming with effects in a functional language therefore quickly becomes laborious. Additionally, the functions above are adapted specifically to our logging effect; using any other effect would require a different implementation. Therefore, we should abstract over the effects in the computation.

This abstraction can be found in the form of *monads* (Peyton Jones and Wadler 1993; Wadler 1992). For our purpose, a monad represents a computation with some effect. It is a type constructor that takes the return type of the computation as a parameter. For a type constructor to be a monad, it needs to define two functions: **return** and `>>=` (pronounced “bind”). The **return** function wraps a value in the monad and `>>=` sequences 2 monadic computations. In Koka, we cannot call these functions **return** and `>>=`, so we call them **pure** and **bind**, respectively.

```

1  alias log<a> = (v: a, msg: string)
2
3  fun pure(v: a): log<a>
4    (v, "")
5
6  fun bind(m: log<a>, k: a -> log<b>): log<b>
7    val (a, s) = m
8    val (b, s') = k(a)
9    (b, s ++ s')
10
11 fun log(msg: string): log<()>
12  ((, msg)

```

The `is_admin` and `all_admins` can then be written using these functions instead of dealing with the strings in the tuples directly. If we then create another monadic effect, we can then change which effect `is_admin` uses without changing the implementation of `all_admins`.

```

1  fun all_admins(list)
2    match list
3      Nil() -> pure(True)
4      Cons(x, xs) -> is_admin(x).bind fn(y)
5                    all_admins(xs).bind fn(ys)
6                    pure(y && ys)
7
8  fun is_admin(user_id: int): log<bool>
9    if admins.any(fn(x) x == user_id)
10   then pure(True)
11   else
12     log("Denied " ++ show(user_id) ++ "\n").bind fn(())
13     pure(False)

```

In fairness, Koka is not designed for monadic operations and other languages provide more convenient syntax for monads. However, the structure of this program in, for example, Haskell would be roughly the same.

A limitation of the monad approach becomes apparent when we want to use multiple effects. The problem is that the composition of two monads does not yield a monad. This limitation can be worked around with *monad transformers*. A monad transformer takes a monad and adds operations to it. The operations of every effect then need to be implemented on every transformer. Adding a single effect therefore requires additional implementations of its operations every other monad transformer. The number of implementations therefore grows quadratically with the number of effects.

To overcome these limitations, we instead turn to the theory of *algebraic effects*, which allows effects to be defined modularly. In this theory, an effect consists of a set of *effect operations*, which form an interface for the effect. A computation using these operations then needs to be wrapped in a *handler*, which defines the semantics for the operations. Different handlers can then give different semantics to the effect.

We can adapt our program to this model, because Koka natively supports algebraic effects and handlers. In the listing below, we first declare the algebraic effect `log`. This effect has a single operation also called `log`. Then, we define a handler `hLog` for the `log` effect. The handler transforms the effectful computation into a tuple of the log and the return value. Note that the `return` branch matches the `pure` function and that the `log` branch combines the

`bind` and `log` functions from the monadic version. The `is_admin` and `all_admin` functions declare that they use the `log` effect with the *effect row* `<log>`. Additionally, Koka's `map` function is polymorphic over effects, so it can take effectful functions as an argument.

```
1 effect log
2   ctl log(msg: string): ()
3
4 val hLog = handler
5   return(x) (x, "")
6   ctl log(msg)
7     val (x, msg') = resume(())
8     (x, msg ++ msg')
9
10 fun is_admin(user_id: int): <log> bool
11   val result = admins.any(fn(x) x == user_id)
12   if !result then
13     log("Denied " ++ show(user_id) ++ "\n")
14   result
15
16 fun all_admins(l): <log> bool
17   l.map(is_admin).foldl(True, (&&))
18
19 val (result, log) = hLog { [1,2,3,4].all(is_admin) }
```

This implementation looks very much like imperative code, but the type system still resembles the type system of functional languages. Effects are handled in a structured way, but are still convenient to use. There are several other advantages too. The effects are modular and can be combined easily. Additionally, the handlers are modular; any handler can be swapped out for another handler, changing the semantics of the effect. For example, we could write a handler that ignores all `log` calls or stores the logged messages in a list.

Koka is not the only language with support for algebraic effects; other notable examples include Frank (Lindley, McBride, and McLaughlin 2017), Effekt (Brachthäuser, Schuster, and Ostermann 2020), Eff (Bauer and Pretnar 2015), Helium (Biernacki et al. 2019), and OCaml (Sivaramakrishnan et al. 2021).

However, many *higher-order operations*, which are operations that take effectful computations as parameters, cannot be expressed as algebraic effects. Therefore, it is not possible to write modular handlers for these operations. This is known as the *modularity problem* for higher-order effects (Wu, Schrijvers, and Hinze 2014). Several extensions to algebraic effects have been proposed to accommodate for higher-order effects (van den Berg et al. 2021; Wu, Schrijvers, and Hinze 2014). One such extension is the theory of *hefty algebras* by Bach Poulsen and van der Rest (2023), which introduces elaborations to modularly define higher-order effects. Elaborations give semantics to higher-order effects by transforming them into computations with only algebraic effects. This means that evaluation of a computation becomes a two-step process: a computation with higher-order effects is first elaborated into a computation with only algebraic effects, which can then be handled. Like handlers, elaborations are modular, and it is possible to define multiple elaborations for a single effect.

Therefore, there currently exist languages with algebraic effects and there is a theory for hefty algebras, but there are no languages based on hefty algebras yet. This is the gap in research that this thesis aims to fill. The question we aim to answer is:

**How can we design a language with native support for higher-order effects and elaborations based on hefty algebras?**

---

In this thesis, we introduce a novel programming language called *Elaine*. This language features elaborations and higher-order effects as native constructs. This brings the theory of hefty algebras into practice. We also present a set of example programs written in Elaine to demonstrate the use of elaborations as a language feature.

Like handlers for algebraic effects, elaborations require the programmer to specify which elaboration should be applied. In practice, we find that the set of elaborations to apply is often unambiguous. Therefore, we introduce *implicit elaboration resolution*, a type-directed procedure that infers the set of elaborations to apply from context.

We provide a specification for Elaine, including its syntax definition, typing judgments, and reduction semantics. Along with this specification, we provide a prototype implementing this specification written in Haskell in the artifact accompanying this thesis. This implementation includes a parser, type checker, interpreter, pretty printer, and the transformations mentioned above.

**Contributions** The main contribution of this thesis is the specification and implementation of Elaine. This consists of several parts.

- We present a language with both handlers and elaborations based on hefty algebras (Chapter 4). The language specification consists of a syntax definition, typing judgments and reduction semantics (Chapter 6). We conjecture that hefty algebras are the semantics of the language, but we do not prove this. We also generalize effect rows to higher-order effects (Section 6.5.1). The type system of the language is based on these generalized effect rows.
- We provide a prototype for the language with a parser, a type checker based on Hindley-Milner type inference and an interpreter for this language implemented in Haskell. This prototype is available in the artifact.
- We provide a set of examples for programming with higher-order effect operations in Elaine (Appendix A). These examples provide evidence that elaborations are a suitable and convenient construct for writing programs with higher-order effects. These examples can be executed and produce the expected results with the prototype. The examples include the reader effect, the writer effect, structured logging and parser combinators, all of which are implemented using higher-order effects.
- We provide a type-directed procedure for inferring elaborations, which alleviates the programmer from specifying which elaboration should be applied when it can be inferred from the context (Chapter 5).

This thesis consists of the following parts. First, we give an overview of the relevant theory of algebraic effects in Chapter 2 and higher-order effects, including hefty algebras, in Chapter 3. Then, we present Elaine in Chapter 4. Implicit elaboration resolution is discussed in Chapter 5, followed by the specification of Elaine in Chapter 6. Finally, we discuss related work in Chapter 7 and conclude in Chapter 8. Appendix A contains additional examples of Elaine programs.

**Artifact** The artifact accompanying this thesis contains a full prototype implementation for Elaine, written in Haskell. The `README.md` file contains instructions for building and executing the interpreter.

The source code of the parser, type checker, interpreter, and other aspects of the implementation can be found in the `src/Elaine` directory. The `examples` directory contains various example programs written in Elaine, including implementations of the reader effect, writer effect, exception effect, structured logging, and a set of parser combinators.

The artifact is available online at <https://github.com/tertsdiepraam/elaine>.



## Chapter 2

---

# Algebraic Effects

The theory of algebraic effects makes working with effectful operations easier in two ways: it makes effects composable and separates the use and interpretation of effects, which provides a high degree of modularity. Crucially, however, it does not cover many higher-order operations; effect operations that take effectful computations as arguments. The theory of hefty algebras extends the theory of algebraic effects with elaborations for higher-order effects. Since Elaine is based on the theory of hefty algebras, the theory of algebraic effects also applies to Elaine. In this chapter, we give an introduction to algebraic effects. In the next chapter, we discuss its limitations regarding higher-order effects and describe how hefty algebras overcome those limitations.

### 2.1 Monads

We will build up the notion of algebraic effects from monads. Monads were introduced as an abstraction for effectful computation by Moggi (1989a,b). Shortly afterwards, monads were popularized as a technique for writing effectful programs in pure languages (Peyton Jones and Wadler 1993; Wadler 1992).

Many definitions of monads can be given, and various analogies can be employed in explaining them. However, for our purposes, a monad is a type constructor with two associated functions: **return** and `>>=`, with the latter pronounced “bind”. In Haskell, this concept is easily encoded in a type class.

```
1 class Monad m where
2   return :: a -> m a
3   (>>=)  :: m a -> (a -> m b) -> m b
```

Haskell

This type class dictates that **return** constructs a value of type `m a` for any monad `m` and any type `a`. This represents a computation that has no further effects and just “returns” a value. Additionally, we can compose two monadic computations using `>>=`, which takes a monadic computation and a *continuation*. The continuation is the function that should be called after the operation has been performed. The continuation is passed the return value of the operation. Intuitively, the `>>=` operator therefore sequences two computations.

To explain how effectful operations can be encoded with this, we can look at a simple example: the **Maybe** monad. Our goal with this monad is to create an “abort” effect, where the computation stops and returns immediately once the **abort** operation is used, without evaluating the rest of the computation. While the returned value is wrapped in **Just**, the computation continues, but once a computation returns **Nothing**, the computation should abort. So, we define the **abort** operation as **Nothing**.



```

1 data Maybe a
2   = Just a
3   | Nothing
4
5 class Monad Maybe where
6   return = Just
7
8   Just a  >>= k = k a
9   Nothing >>= k = Nothing
10
11 abort :: Maybe a
12 abort = Nothing

```

With this definition, we can chain functions returning **Maybe**. For example, we can define a **head** function with the type `[a] -> Maybe a` that returns the first element of a list if it is non-empty and **Nothing** otherwise. We also define a division function which checks that the divisor is non-zero. These functions can then be composed using the `>>=` operator, such that the composed computation returns **Nothing** if one of these functions returns **Nothing**.

```

1 head :: [a] -> Maybe a
2 head (x:xs) = Just x
3 head _ = Nothing
4
5 safeDiv :: Int -> Int -> Maybe Int
6 safeDiv _ 0 = Nothing
7 safeDiv x y = Just $ div x y
8
9 main = do
10   print $ head [] >>= safeDiv 10 -- -> Nothing
11   print $ head [0,1,2] >>= safeDiv 10 -- -> Nothing
12   print $ head [2,3,4] >>= safeDiv 10 -- -> Just 5

```

A more involved example is the **State** monad. A function that modifies state needs to take the current state and return the new state. For example, a function that takes `a`, returns `b`, and modifies state `s` needs to have the type `a -> s -> (s, b)`. The new state then needs to be passed to the next function that modifies state.

```

1 -- Increment the state by `a` and return the old state
2 inc :: Int -> Int -> (Int, Int)
3 inc a s = (s + a, s)
4
5 multipleIncs :: Int -> (Int, Int)
6 multipleIncs s = let
7   (s', _) = inc 5 s
8   (s'', _) = inc 6 s'
9   in inc 7 s''

```

The types of all functions using state now end with the same pattern: `s -> (s, b)`. This is an opportunity for abstraction, because we can define a type for that pattern. Since this type represents the state effect, this type is called **State**.

```

1 newtype State s a = State (s -> (s, a))
2
3 -- so that the inc function becomes:
4 inc :: Int -> State Int Int
5 inc a = State (\s -> (s + a, s))

```

Haskell

This `State s` type becomes a monad if we implement `return` and `>>=` for it. This allows us to compose functions returning the `State` type. Additionally, we define the `get` and `put` functions as a basic interface for interacting with state.

```

1 instance Monad (State s) where
2   return x = State (\s -> (s, x))
3
4   State fa >>= k = State $ \s ->
5     let (s', a) = fa s
6         State fb = k a
7     in fb s'
8
9   get :: State s s
10  get = State (\s -> (s, s))
11
12  put :: s -> State s ()
13  put s = State (\_ -> (s, ()))

```

Haskell

To retrieve the final value of the computation, we define a function `runState`, which takes an initial state and returns a pair of the final state and the returned value.

```

1 runState :: s -> State s a -> (s, a)
2 runState initialState (State func) = func initialState

```

Haskell

The `inc` operations can then be sequenced using the `>>=` operator. Because the return value of `inc` is irrelevant in the computation, we define a shorthand operator `>>`, which ignores the return value of the first operation.

```

1 (>>) :: Monad m => m a -> m b -> m b
2 a >> b = a >>= \_ -> b
3
4 inc :: Int -> State Int Int
5 inc x = get >>= \s -> put (s + x) >>= return s
6
7 multipleIncs :: State Int Int
8 multipleIncs = inc 5 >> inc 6 >> inc 7
9
10 main = print (runState 0 bar) -- prints 0 + 5 + 6 + 7 = 18

```

Haskell

This is the power of monads: they allow us to abstract over the effectful operations, while also declaring the effects that a function requires in its return type. In the final example, we do not have to think about how the `State` monad works any more, but only use the `get` and `put` operations to build complex computations. The abstraction separates the interface from the implementation. This modularity is one of the core motivations of the study of effects.

To make working with monads more convenient, Haskell also features `do`-notation, which is syntactic sugar for the `>>=` and `>>` operators. Using `do`-notation, the `multipleIncs` computation from the previous example can be written as:

```

1 | multipleIncs = do
2 |   inc 5
3 |   inc 6
4 |   inc 7

```

If the results from the `inc` computations needs to be used, the `<-` operator, which is part of `do`-notation, can be used to bind the result of a computation to a variable. For example, the sum of all the results from the `inc` calls can be returned.

```

1 | multipleIncs' = do
2 |   a <- inc 5
3 |   b <- inc 6
4 |   c <- inc 7
5 |   return (a + b + c)
6 |
7 | -- which is equivalent to
8 | multipleIncs' =
9 |   inc 5 >>= \a ->
10 |     inc 6 >>= \b ->
11 |       inc 7 >>= \c ->
12 |         return (a + b + c)

```

Monads are a convenient model for programming with effects in Haskell, while also staying true to its functional paradigm. However, monads are also limited, since they cannot be composed. Imagine, for instance, a computation that decrements some state and returns the new value, but also asserts that the value never becomes negative and returns **Nothing** in that case. This computation might look as follows.

```

1 | decrement :: State Int (Maybe Int)
2 | decrement = get >>= \s ->
3 |             if s > 0
4 |             then put (s - 1) >> return (return (s - 1))
5 |             else return abort

```

The `abort` effect has to be wrapped in additional `return` calls and does not interact with the state. The type `State Int (Maybe Int)` is not a combined monad for both effects, but one monad wrapped in another. Instead, we need some combined monad `MaybeState` that combines the operations of both monads.

```

1 | decrement :: MaybeState Int Int
2 | decrement = get >>= \s ->
3 |             if s > 0
4 |             then put (s - 1) >> return (s - 1)
5 |             else abort

```

While it is possible to define such a monad, we would need to define one for every combination of monad operations that we would like to use, which quickly becomes cumbersome. Hence,

we need to look elsewhere for a solution. One solution to this is to use monad transformers, as explained in Section 7.1. Another solution is to compose effects in the *free monad*.

## 2.2 Composing Effects with the Free Monad

The free monad is a monad that encodes the structure of a program without imposing semantics (Swierstra 2008). The free monad takes a functor `f` as an argument. The free monad then gives a syntactic description of the operations given by that functor. It is therefore the trivial monad parametrized by the operations of `f`. In Haskell, the free monad is implemented as the `Free` data type. The definitions in this section are based on the implementation by Kammar, Lindley, and Oury (2013) and the encoding of the free monad by Bach Poulsen and van der Rest (2023).

```

1 data Free f a
2   = Pure a
3   | Op (f (Free f a))
4
5 instance Functor f => Monad (Free f) where
6   return = Pure
7   Pure x >>= f = f x
8   Op g >>= f = Op (fmap (>>= f) g)

```

This type class states that, if `f` is some functor, then `Free f` is monad. For example, if we make some functor `State s`, then `Free (State s)` is a monad. Of course, this is only useful if `Free (State s)` is a monad with the same functionality as the original state monad. These functors are defined with a constructor for each effect operation. Since the free monad encodes the syntax of the effect operations, we define the `State` functor with constructors corresponding to the operations of the state effect: `get` and `put`. This is a functor over the `k` parameter, which represents the *continuation*. Note that we do not have to give definitions of `return` and `>>=` since those are defined generically for any `f` on `Free`. We only have to derive the default `Functor` instance.

```

1 data State s k = Put s k | Get (s -> k)
2   deriving Functor

```

Similarly, we can write the abort effect as a free monad, which we previously implemented using the `Maybe` monad. Recall that there is only one operation for this effect: `abort`. Hence, the `Abort` functor only needs a single constructor, which we also call `Abort`. The `Abort` constructor does not use the continuation because it signals that evaluation should stop.

```

1 data Abort k = Abort
2   deriving Functor

```

In contrast with monads, these functors can be meaningfully composed. We define a type-level operator `+`, which represents a coproduct for functors. This operator can be thought of as `Either` for functors, since `Either` is the coproduct for types. We use this operator to build lists of functors. Just like lists have a `Cons` and `Nil`, these lists consist of `+` and `End`, where `End` is a functor without any constructors.

```

1 infixr 6 +
2 data (f + g) a = L (f a) | R (g a)
3   deriving Functor
4
5 data End k
6   deriving Functor

```

The `End` functor has the property that it does not add any operations. Therefore, we have that `Free (f + End)` is functionally the same as `Free f` and that `Free End a` is equivalent to `a`. We can then make monads for any combination of the functors we have defined, such as `Free (State s + End)`, `Free (Abort + End)`, or `Free (State s + Abort + End)`. In general, we can construct a monad for any set of functors.

However, we have no way to use any of the effect operations. For example, if we have the monad `Free (State s + Abort + End)`, how would we use the `get` operation that we expect from the state monad? The solution is to give a definition for `get` for the free monad if and only if `State` is one of the composed functors. We do this with a type class relation `<`, which defines an injection from a functor `f` to any composed functor `g` that contains `f`. We can use this injection to define the `get`, `put`, and `abort` functions. These functions are called *smart constructors*.

```

1 class f < g where
2   inj :: f k -> g k
3
4 instance f < f where inj = id
5 instance f < (f + g) where inj = L
6 instance f < h => f < (g + h) where inj = R . inj
7
8 get :: State s < f => Free f s
9 get = Op $ inj $ Get Pure
10
11 put  :: State s < f => s -> Free f ()
12 put s = Op $ inj $ Put $ Pure ()
13
14 abort :: Abort < f => Free f ()
15 abort = Op $ inj $ Abort

```

This makes it possible to construct a computation using all those operations. For example, a computation that checks the state, asserts that it is larger than 0, and then decrements the state by 1.

```

1 decrement :: Free (State Int + Abort + End) Int
2 decrement = get >>= \s ->
3   if s > 0
4   then put (s - 1) >>= return (s - 1)
5   else abort

```

To evaluate this computation, we need to define an algebra for the `Free` data type. Conceptually, the evaluation should remove the effects one by one from the monad. To do that, there needs to be a function with the type

$$\text{Free } (f + f') \text{ a} \rightarrow \text{Free } f' \text{ b}$$

for every `f` and finally a function `Free End a -> a` to reduce the free monad to a final value. Following Plotkin and Pretnar (2009), these functions are called *handlers*. In general, any handler operating on `Free` needs to take two cases into account, since `Free` has two constructors: `Pure` and `Op`. By using a fold over `Free` we can define a handler in terms of two smaller functions, corresponding to these two constructors:

```

1 | fold :: Functor f => (a -> b) -> (f b -> b) -> Free f a -> b
2 | fold gen _ (Pure x) = gen x
3 | fold gen alg (Op f)  = alg (fmap (fold gen alg) f)
4 |
5 | -- the return case
6 | ret :: a -> Free f' b
7 | -- and the case for handling operations
8 | op  :: f (Free f' b) -> Free f' b

```

However, to generalize the handler, we add a parameter `p` as well. This parameter can be used by the handlers to thread some value through the computation. For some effects, this parameter will simply be the unit value `()`, but for the state effect, this parameter will represent the state. With this parameter, the `handle` function, which constructs handlers, is defined as follows:

```

1 | handle :: (Functor f, Functor f')
2 |         -- a function for the return case:
3 |         => (a -> p -> Free f' b)
4 |         -- a function for handling operations:
5 |         -> (f (p -> Free f' b) -> p -> Free f' b)
6 |         -- the type of the resulting handler:
7 |         -> Free (f + f') a -> p -> Free f' b
8 | handle ret f = fold ret $
9 |   \case
10 |     L x -> f x
11 |     R x -> \p -> Op $ fmap (\m -> m p) x

```

Handlers for the various effects can then be constructed using `handle`. For each computation, we need a handler for each effect to remove them from the free monad so that only the `End` functor remains.

```

1 | handleAbort :: Functor f => Free (Abort + f) a -> Free f (Maybe a)
2 | handleAbort c = handle
3 |   (\a _ -> return $ Just a)
4 |   (\Abort () -> return Nothing)
5 |   c ()
6 |
7 | handleState :: Functor f => s -> Free (State s + f) a -> Free f (s, a)
8 | handleState = flip $ handle
9 |   (\x s -> return (s, x))
10 |   (\x s -> case x of
11 |     Put s' k -> k s'
12 |     Get k -> k s s)

```

Applying the handlers for all effects leaves us with `Free End a`, which should be reduced to `a`. This is done with the `handleEnd` function.

```

1 | -- The Op case does not need to be handled since End
2 | -- cannot be constructed
3 | handleEnd :: Free End a -> a
4 | handleEnd (Pure a) = a

```

Haskell

Finally, we can reduce a free monad with multiple effects to a single result by applying all handlers for the effects. So, the `decrement` function above requires handlers for `State s`, `Abort`, and `End` and can be evaluated as follows:

```

1 | result :: (Int, Maybe Int)
2 | result = handleEnd $ handleState (0::Int) $ handleAbort decrement

```

Haskell

With this encoding, we can use the abort and state effects together, while providing separate handlers per effect. Note that the order in which the handlers are applied matters for the return type of `result`. If abort is handled first and state second, the final type is `(Int, Maybe Int)`. If state is handled first, it is `Maybe (Int, Int)`.

While the plumbing needed for a free monad is extensive, it has many advantages over regular monads. First, we can combine multiple functors in our type signatures. Second, we can define operations that work for any effect composition that contains an effect. Third, we can provide modular handlers that handle a single effect from the composed functors. If all effects are defined in this way, then effect is automatically compatible with all other effects.

Finally, we have not only gained modularity for the effects themselves, but also for the handlers. The effects have become an interface, while the handlers provide the semantics. Within this framework, the semantics of effects can be changed without touching the type and definition of the computation. There is nothing preventing different implementations of the handlers. It is, for example, possible to define a state handler in which `put` operations are ignored, keeping the state is constant.

## 2.3 Algebraic Effects

The free monad encoding in the previous section is an implementation of algebraic effects in Haskell. The term “algebraic” comes from the fact that this method works for effects that can be described as algebraic theories (Plotkin and Power 2001). Later, Plotkin and Power (2003) showed that this is only possible for effects that satisfy the *algebraicity property*.

The algebraicity property states that the `>>=` operation distributes over the computation parameters of an operation. This means that if there is some operation `op` that has some parameter of type `k` then the following computations should be equivalent for some continuation `k'`:

$$(\text{op } k) \gg= k' \quad == \quad \text{op } (k \gg= k')$$

So, if the state effect satisfies the algebraicity property, the following equality should hold for any continuations `k` and `k'`:

$$(\text{Op } (\text{Put } s \ k)) \gg= k' \quad == \quad \text{Op } (\text{Put } s \ (k \gg= k'))$$

Intuitively, this matches how we expect state to behave: if the state is changed, it will remain changed throughout the rest of the continuation, until it is changed again.

By construction, the algebraicity property holds for any effect we have defined in the previous section. This can be derived from the definitions of `>>=` on `Free` and `fmap` for the effects. Indeed, we can apply the definitions to `Free (State s)` to verify that the algebraicity property holds.

```

1 | Op (Put s k) >>= k'
2 | -- apply >>= of Free:
3 | Op (fmap (>>= k) (Put s k'))
4 | -- apply fmap of Put:
5 | Op (Put s (k >>= k'))

```

Haskell

Consequently, any effect that does not satisfy the algebraicity property cannot be written as the free monad. While the state and abort effects satisfy this property, *higher-order effects*, do not. Examples of higher-order effects are exception catching and the reader effect with a local operation. Those effects are discussed in Chapter 3.

## 2.4 Building a Language with Algebraic Effects

Although the previous sections contain an encoding of algebraic effects, there are details in this encoding that we might like to hide. For instance, writing all return types as `Free f a` for every function becomes repetitive. Every returned value also needs to be wrapped in `pure` to be mapped to the monad. Our goal is then to remove as much of the additional syntax that is required to work with effects when compared to pure functions.

This is where we reach the limits of what we can achieve with the encoding of the free monad in Haskell. If we instead design a new language which integrates algebraic effects as a core feature of the language, we have much more freedom in designing a syntax and semantics that work well for this purpose.

Elaine is a language with support for algebraic effects, but it also supports higher-order effects. Therefore, this section focuses on Koka (Leijen 2014, 2023), which only supports algebraic effects. Since Elaine is heavily inspired by Koka, the same concepts apply to Elaine.

At the core of such languages lies the following concept: all functions implicitly return the free monad with some effects. Therefore, we write  $a \rightarrow e\ b$ , which is analogous to  $a \rightarrow \text{Free } e\ b$  in the Haskell encoding. In that signature, we call  $e$  the *effect row* and its elements as *effects*. So, the function signature  $a \rightarrow e\ b$  should be read as: “this function takes an  $a$  and returns  $b$  with effect row  $e$ .”

Instead of using type-level operators, we can introduce special syntax for effect rows, too. Following Koka, we will write effect rows as

$$\langle e_1, e_2, \dots, e_n \rangle.$$

Additionally, effects becomes a special construct separate from monads and functors. Therefore, effect rows can get special treatment in the type system. It should be able to, for example, reason about equality between effect rows with the same effects in different orders, such as  $\langle a, b \rangle$  and  $\langle b, a \rangle$ . In the type system, we are then allowed to use different orders of effects interchangeably. We can reason about inclusion of effects in an effect row with the following notation:

$$\langle e_1, e_2, \dots, e_n \mid e_s \rangle,$$

where  $e_s$  is the tail of the effect row; a variable representing the effect row with which this effect row can be extended. For example, the effect row  $\langle \text{abort} \mid e \rangle$  consists of the `abort` effect and some other effect row  $e$ . We can define the same effects as before, such as the state and abort effects, but in Koka, we do not define them as functors. Instead, we define them using the **effect** keyword and each constructor of the functors is then declared with the `ctl` keyword.



```

1 effect abort
2   ctl abort(): a
3
4 effect state<a>
5   ctl get(): a
6   ctl put(x: a): ()

```

Koka

The equivalent of `Free (State s + Abort + End) a` is then `<state<s>,abort> a`. The equivalent of a handler would then be a function which takes `() -> <f|e> a` and returns `<|e> a`. In Koka, such a function can be defined with the **handler** construct, which requires an implementation for each operation of an effect and a special function for the return case.

Note the similarity to the `handle` function we defined in Haskell before. In the case of abort effect, this handler is assigned to variable for later use. The state handler resembles the original state monad: it takes a computation `() -> <state<s>|e> a` and returns `s -> e (s, a)`. For example, the return arm yields the anonymous function `fn(x) (s, x)`. The continuation can be called in a handler with the `resume` function in Koka. Since handling is defined by a fold, just like in the Haskell encoding, the effect is already handled in the continuation. For the state effect, `resume` therefore returns a function with the type `s -> e (s, a)`.

```

1 val hAbort = handler
2   return(x)   Just(x)
3   ctl abort() Nothing
4
5 fun hState(init, c)
6   fun h(c')
7     with handler
8       return(x) fn(s) (s, x)
9       ctl get() fn(s) resume(s)(s)
10      ctl put(n) fn(_) resume(())(n)
11      c'()
12
13   h(c)(init)

```

Koka

In the free monad encoding in Haskell, the state had to be passed through the handlers as a parameter. Koka is a bit more flexible and allows us to return values with effectful computations. Therefore, it does not need the additional parameter.

Koka helps us hide some details, but the structure in the listing above is largely the same as with the free monad encoding. The larger differences become apparent when we want to use the effects in some computation. A part of the decrement function is listed below.

```

1 fun decrement(): <state<int>,abort> int
2   val s = get()
3   if s == 0 then
4     abort()
5
6     put(s - 1)
7     s - 1
8
9 val x = hAbort { hState(3, decrement) } // -> Just(2)
10 val y = hAbort { hState(0, decrement) } // -> Nothing

```

Koka

The `>>=` operator is entirely implicit here. Therefore, it is similar to Haskell's `do`-notation. However, in `do`-notation, every effectful operation needs to be on a separate line. For example, if the state needs to be incremented by 1, this can be achieved in one line in Koka, but in Haskell using `do`-notation requires two lines.

```
1 | put(get() + 1)
```

Koka

```
1 | do
2 |   x <- get
3 |   put (x + 1)
```

Haskell

In Koka, effectful operations can be used anywhere as long as they are wrapped in a corresponding handler. In the end, the syntax is closer to imperative programming languages than functional programming languages. However, the type system still very much resembles that of a functional language. This is important because this means that we have not lost any of the type safety that the monadic treatment of effects provides. The signature of a function in Koka still gives a complete specification of all effects that a function might perform. In imperative languages, this information is entirely missing from the function signature. For example, the type system can assert that a function is entirely pure. In the listing below, the `<>` in the type of the function asserts that it does not require effects, yet the `println` function requires an effect. Hence, Koka's type checker will yield a type error.

```
1 | fun should_be_pure(x: int): <> int
2 |   println("This will give a type error!")
3 |   x + 10
```

Koka

As will become clear in Chapter 4, Elaine takes a lot of inspiration from Koka. Handlers and effects are defined in the same way, modulo some syntactic differences. What sets Elaine apart, is that it also supports higher-order effects, which will be explained in the next chapter.



## Chapter 3

# Higher-Order Effects

In the previous chapter, we explained the concept of algebraic effects. Any effect that satisfies the algebraicity property is algebraic. However, many *higher-order effects* are not algebraic. An effect is higher-order if one of its operations takes effectful computations as parameters (Bach Poulsen and van der Rest 2023). As a result, it is not possible to give modular implementations for these operations using effect handlers, like we can do for algebraic operations. This chapter details the difficulties around higher-order effects and discusses hefty algebras, the theory that Elaine is based on.

### 3.1 Computation Parameters

Recall that an effect in the free monad encoding is a functor over some `k` with some constructors. The type `k` represents the continuation of the computation. Naturally, it is possible to write a constructor with multiple parameters of type `k`. For example, we could make a `Branch` functor which takes a boolean and two computations. Based on the boolean value, it selects the branch to evaluate. It is essentially an **if-else** expression expressed as an effect.

```
1 data Branch k = Branch Bool k k
2
3 branch :: Branch < f => Bool -> Free f a -> Free f a -> Free f a
4 branch b ifTrue ifFalse :: Op $ inj $ Branch b ifTrue ifFalse
```

Haskell

The important observation with this effect is that both `ifTrue` and `ifFalse` behave like continuations. To examine why, consider the following computation.

```
1 branch b (pure 0) (pure 1) >>= \x -> pure (x + 1)
```

Haskell

Like previously established, the `>>=` operator defined by `Free` distributes over the computation parameters. This yields the following expression.

```
1 branch b
2   (pure 0 >>= \x -> pure (x + 1))
3   (pure 1 >>= \x -> pure (x + 1))
4 -- which reduces to
5 branch b (pure 1) (pure 2)
```

Haskell

This computation has the same intended semantics as the original. The distribution of `>>=` therefore does not change the semantics and hence the effect is algebraic. Therefore, there would be no problem encoding this effect in Haskell using the encoding in the previous chapter and, by extension, in Koka.

This is what we mean by saying that the parameters are computation-like: the continuation can be appended to the parameters without changing the semantics of the effect.

## 3.2 Breaking Algebraicity

For other effects, however, the intended semantics are not such that the computation parameters are continuation-like.

One such effect is the `Reader` effect. Traditionally, the `Reader` monad has two operations: `local` and `ask`. The latter functions much like the `get` operation from the state effect and is algebraic. However, the `local` operation is more complex. It takes two parameters, a function `f` and a computation `c`. The intended semantics are then that whenever `ask` is used within `c`, the function `f` is applied to the returned value. To see why the `local` operation breaks algebraicity, consider the following computation.

```
1 | local (* 2) ask >>= \x -> ask >>= \y -> pure x + y
```

`Haskell`

Only the first `ask` operation is inside the `local` operation and should therefore be doubled. If the `Reader` effect was algebraic, we should be able to distribute the `>>=` operator again without changing the semantics of the program. However, doing so yields the following computation.

```
1 | local (* 2) (ask >>= \x -> ask >>= \y -> pure x + y)
```

`Haskell`

Now, both `ask` operations are inside the `local` operation, so both values will be doubled. For example, if we had installed a handler that makes `ask` return 1, the first computation would return  $2 + 1 = 3$  and the second  $2 + 2 = 4$ . Therefore, we have shown with a counterexample that the `Reader` effect cannot be algebraic.

A similar argument holds for the `Except` effect, which also has two operations: `catch` and `throw`. In the simplest form, `throw` resembles the `abort` effect, but it takes a parameter that represents an error message. The `catch` operation evaluates its first parameter and jump to the second if it fails, much like the try-catch constructs of languages with effects. Again, we examine a simple example program to show that `Except` violates algebraicity.

```
1 | catch (pure False) (pure True) >>= throw -- -> throws False
2 | -- then distributing >>= yields
3 | catch
4 |   (pure False >>= throw)
5 |   (pure True >>= throw)
6 | -- which simplifies to
7 | catch (throw False) (throw True)           -- -> throws True
```

`Haskell`

Before distributing the `>>=` operator the computation should throw `False`, but after it should throw `True`. So, again, the semantics have changed by distributing the `>>=` and therefore `Except` is not algebraic.

## 3.3 The Modularity Problem

Taking a step back from effects, defining a function for exception catching is possible. Recall that the `throw` operation is algebraic, therefore, a handler for it can be defined. If we assume some handler for it called `handleThrow` returns an `Either` where `Left` is the value from

`throw` and `Right` is the value from a completed computation, we can define `catch` in terms of that function.

```

1 | catch c1 c2 =
2 |   case handleThrow c1 of
3 |     Left e -> c2
4 |     Right a -> return a

```

The distinction between effects which are and which are not algebraic has been described as the difference between *effect constructors* and *effect destructors* (Plotkin and Power 2003). The `local` and `catch` operations have to act on effectful computations and change the meaning of the effects in that computation. So, they have to deconstruct the effects in their computations using handlers. An imperfect heuristic for whether a function can be an algebraic effect is to check whether the implementation requires a handler. If it uses a handler, it probably cannot be an algebraic effect.

An algebraic effect can have a modular implementation: a computation can be reused in different contexts by using different handlers. For these higher-order effects such as `catch` and `local`, this is not possible. This is known as the *modularity problem* with higher-order effects (Wu, Schrijvers, and Hinze 2014). This is the motivation behind the research on higher-order effects, including this thesis. It is also the problem that the theory of hefty algebras aims to solve.

### 3.4 Hefty Algebras

Several solutions to the modularity problem have been proposed (van den Berg et al. 2021; Wu, Schrijvers, and Hinze 2014). Most recently, Bach Poulsen and van der Rest (2023) introduced a solution called hefty algebras. The idea behind hefty algebras is that there is an additional layer of modularity, specifically for higher-order effects.

For a full treatment of hefty algebras, we refer to Bach Poulsen and van der Rest (2023). In addition, the encoding of hefty algebras is explained in more detail by Bach Poulsen (2023a).

At the core of hefty algebras are hefty trees. A hefty tree is a generalization of the free monad to higher-order functors, which will write `HFunctor`. In the listing below, we also repeat the definition of a functor from the previous chapter for comparison.

```

1 | -- a regular functor
2 | class Functor f where
3 |   fmap :: (a -> b) -> f a -> f b
4 |
5 | -- a higher-order functor
6 | class (forall f. Functor (h f)) => HFunctor h where
7 |   hmap :: (f a -> g a) -> (h f a -> h g a)

```

The definition of a hefty tree, with the free monad for reference, then becomes:

```

1 | -- free monad
2 | data Free f a
3 |   = Pure a
4 |   | Op (f (Free f a))
5 |
6 | -- hefty tree
7 | data Hefty h a
8 |   = Return a
9 |   | Do (h (Hefty h) (Hefty h a))

```

A hefty tree and a free monad are very similar. Like for the free monad, we can define **return** and **>>=** for **Hefty h**, so that it can be used as a monad. Additionally, we define **injH**, **<:**, and **:+:** as analogues for **inj**, **<**, and **+** from the previous chapter, respectively. We refer to Bach Poulsen (2023a) for the definition of these operators. Furthermore, any functor can be lifted to a higher-order functor with a **Lift** data type.

```

1 | data Lift f (m :: * -> *) k = Lift (f k)
2 |   deriving Functor
3 |
4 | instance Functor f => HFunctor (Lift f) where
5 |   hmap _ (Lift x) = Lift x

```

In algebraic effects, the evaluation of a computation can be thought of as a transformation of the free monad to the final result:

$$\text{Free } f \ a \xrightarrow{\text{handle}} \ b.$$

Using hefty algebras, the evaluation instead starts with a *hefty tree*, which is *elaborated* into the free monad. The full evaluation of a computation using hefty algebras then becomes:

$$\text{Hefty } h \ a \xrightarrow{\text{elaborate}} \text{Free } f \ a \xrightarrow{\text{handle}} \ b.$$

This elaboration is a transformation from a hefty tree into the free monad, defined as an algebra over hefty trees. The algebras are then used in **hfold**; a fold over hefty trees.

```

1 | hfold :: HFunctor h
2 |       => (forall a. a -> g a)
3 |       -> (forall a. h g (g a) -> g a)
4 |       -> Hefty h a
5 |       -> g a
6 | hfold gen _ (Return x) = gen x
7 | hfold gen alg (Do x)   =
8 |   alg (fmap (hfold gen alg) (hmap (hfold gen alg) x))
9 |
10 | elab :: HFunctor h
11 |       => (forall a. h (Free f) (Free f a) -> Free f a)
12 |       -> Hefty h a
13 |       -> Free f a
14 | elab elabs = hfold Pure elabs

```

For any algebraic – and thus lifted – effect, this elaboration **eLift** is trivially defined by unwrapping the **Lift** constructor. Applying **eLift** to **elab** then gives a function which elaborates **Hefty (Lift f) a** to **Free f a** for any functor **f**.

```

1 | eLift :: g < f => Lift g (Free f) (Free f a) -> Free f a
2 | eLift (Lift x) = Op (inj x)

```

Haskell

To elaborate multiple effects, elaborations can be composed using the `^` operator as defined by Bach Poulsen and van der Rest (2023). The composed elaborations are then applied all at once, because an elaboration of a `Hefty h a` returns a `Free f a`, which cannot contain any higher-order effects.

Using elaborations, we can modularly define higher-order operations, such as `catch` and `local` operations. For example, the `Except` effect can be elaborated to a computation using only the algebraic `ThrowA` effect. This elaboration is given in Figure 3.1.<sup>1</sup>

For the `Reader` effect, the `local` operation needs to be elaborated, but the `ask` operation can be handled. So, we define an algebraic effect `AskA` with only the `ask` operation. This effect is handled by `handleAskA`. The implementation of both the handler and elaboration are listed in Figure 3.2.

The `catch` and `local` operations in these examples are modular, because the elaborations can be swapped out for different elaborations. Additionally, we can write computations using both effects. For example, the elaboration of a computation with both a `catch` and `reader` effect looks as follows:

```

1 | main = handleThrow $ handleAsk 0 $
2 |   eLab (eCatch ^ eReader ^ eLift) c

```

Haskell

Similar to how Koka is based on the theory of algebraic effects, Elaine is based on hefty algebras. Therefore, Elaine supports elaborations for higher-order effects. While the intended semantics for expressions in Elaine are hefty trees, the semantics of Elaine's `eLab` construct are currently unclear, as discussed in Section 6.5.1.

<sup>1</sup>This code is simplified and does not compile in this form. In reality the result of the handler needs to be wrapped in a function `hup`, which reorders the effects in the effect row to match the context. The same goes for the example for the reader effect below. This `hup` requires some additional machinery as explained by Bach Poulsen (2023b).



```

1  data ThrowA e k = ThrowA e
2    deriving Functor
3
4  handleThrowA :: Functor f
5                => Free (Abort + f) a -> Free f (Maybe a)
6  handleThrowA c = handle
7    (\a _ -> Pure $ Right a)
8    (\(Throw e) () -> Pure $ Left e)
9    c ()
10
11 data Except e f k
12   = Throw e
13   | Catch (f a) (f a) (a -> k)
14
15 deriving instance Functor (Except e f)
16
17 instance HFunctor (Except e) where
18   hmap _ (Throw x) = Throw x
19   hmap f (Catch m1 m2 k) = Catch (f m1) (f m2) k
20
21 throw :: Except e <: h => e -> Hefty h a
22 throw x = Op (injH (Throw x))
23
24 catch :: Except <: h
25         => Hefty h a -> Hefty h a -> Hefty h a
26 catch m1 m2 = Do $ inj $ Catch m1 m2 Return
27
28 eExcept :: ThrowA e < f
29         => Except e (Free f) (Free f a)
30         -> Free f a
31 eExcept (Throw e) = Do (inj (ThrowA e))
32 eExcept (Catch m1 m2 k) =
33   handleThrowA c1 >>= \v ->
34     case v of
35       Left e -> c2 >>= k
36       Right x -> k x
37
38 comp :: Hefty (Except e :+: Lift (ThrowA e)) Int
39 comp = catch (throw) (return 4)
40
41 result = handleAbort $ elab (eExcept ^ eLift) comp

```

Figure 3.1: Definition, elaboration and use of the Except effect.

```

1  data AskA r k = AskA (r -> k)
2    deriving Functor
3
4  handleAskA :: Functor f
5              => r -> Free (AskA + f) a -> Free f a
6  handleAskA v m =
7    handle
8      (\a () -> Pure a)
9      (\(Ask k) -> k v)
10     m ()
11
12 data Reader r m k =
13   = Ask (r -> k)
14   | forall a. Local (r -> r) (m a) (a -> k)
15
16 deriving instance Functor (Reader r m)
17
18 instance HFunctor (Reader r) where
19   hmap _ (Ask k) = (Ask k)
20   hmap f (Local f m k) = Local g (f m) k
21
22 ask :: Reader r <: h
23     => Hefty h r
24 ask = Do $ injH $ Ask Pure
25
26 local :: Reader r <: h
27        => (r -> r) -> Hefty h a -> Hefty h a
28 local f m = Do $ injH $ Local f
29
30 eReader :: AskA r < f
31          => Reader r (Free f) (Free f a)
32          -> Free f a
33 eReader (Ask k) = Op $ inj $ AskA k
34 eReader (Local f m k) =
35   askA >>= \r ->
36     handleAskA (f r) m >>=
37       k
38
39 comp :: Hefty (Reader Int :+: Lift Abort) Int
40 comp = ask >>= \x ->
41         local (* 2) (ask >>= \y -> return (x + y))
42
43 result = handleAbort $ elab (eReader ^ eLift) comp

```

Figure 3.2: Definition, elaboration and use of the Reader effect.



# Chapter 4

---

## A Tour of Elaine

The language designed for this thesis is called “Elaine”. The distinguishing feature of this language is its support for higher-order effects via elaborations. The basic feature of elaborations has been extended with implicit elaboration resolution, which is detailed in Chapter 5.

### 4.1 Overview

At its core, Elaine is based on the lambda calculus, extended with algebraic and higher-order effects. The feature set has been chosen to be comprehensive enough for fairly extensive programs, which are given in Appendix A.

Elaine’s syntax is mostly inspired by Koka (Leijen 2014, 2017) and Rust (Matsakis and Klock 2014). The keywords of the language will be particularly familiar to Rust programmers. It is designed to be relatively simple to parse, which is most clearly reflected in the fact that whitespace is ignored and that there are no infix operators. Elaine requires semicolons at the end of each statement and requires computations consisting of multiple statements to be wrapped in braces.

All expressions in Elaine are statically and strongly typed with a type system based on Hindley-Milner style type inference (Hindley 1969; Milner 1978). The type system has a special treatment for effect rows similar to Koka’s approach. In most cases, types can be completely inferred and do not need to be specified. Additionally, algebraic data types and tuples are supported for modelling complex data.

Like Koka, Elaine has strict semantics. This means that effects can only occur during function application (Leijen 2014). Additionally, the order in which effects are performed is very clear in this model. We believe this makes effects easier to reason about than in a language with lazy evaluation. Naturally, lazy evaluation can still be encoded into a strict language (Wadler 1996). It also matches the more imperative style Elaine programs are written in. There is currently no way for Elaine programs to interact with the operating system; there is no equivalent to the `IO` monad from Haskell or the `console` and `fsys` effects from Koka.

The source code for the Elaine prototype and additional examples are included in the artifact accompanying this thesis. The full specification for Elaine, including typing judgements and reduction semantics are given in Chapter 6.

### 4.2 Basics

As is tradition with introductions to programming languages, we start with a program that shows the string `"Hello, world!"`.

```
1 | # The value bound to main is the return value of the program
2 | let main = "Hello, world!";
```

Elaine

This example highlights several aspects of Elaine. Comments start with `#` and continue until the end of the line. We bind variables with the `let` keyword. The `main` variable is required, and the value assigned to it is printed at the end of execution. In contrast with other languages, `main` is not a function in Elaine. Note that statements are required to end with a semicolon.

In addition to strings, Elaine features integers and booleans as built-in types. To operate on these types, we need functions to perform the operations. By default, there are no functions in scope, however, we can bring them in scope by importing the functions from the `std` module with the `use` keyword. For example, we can write a program that computes  $5 \cdot 2 + 3$ :

```
1 | use std;
2 | let main = add(mul(5, 2), 3);
```

Elaine

The `std` module contains functions for boolean and integer arithmetic, comparison of values, and more. The full list of functions is given in Section 6.6. To show off some more functions, below is a program that compares the results of two calculations. Note that `-` is allowed as part of an integer literal, but not as an operator. The functions used here are “greater than” (`gt`), exponentiation (`pow`), negation (`neg`), and multiplication (`mul`).

```
1 | use std;
2 | let main = gt(
3 |     pow(2, 5),
4 |     neg(mul(25, -30)),
5 | );
```

Elaine

Let-bindings can be used to split up a computation, both at the top-level and within braces, which are used to group sequential expressions. Like in Rust, a sequence of expressions evaluates to the last expression. Expressions are only allowed to contain variables that have been defined above the expression, so the order of bindings is significant. This rule also disallows recursion. Below is the same comparison written with some bindings.

```
1 | use std;
2 | let a = pow(2, 5);
3 | let main = {
4 |     let b = mul(25, -30);
5 |     gt(a, neg(b))
6 | };
```

Elaine

Functions are defined with `fn`, followed by a list of arguments and a function body. Unlike Haskell, functions are called with parentheses. Note that Elaine does not support function currying.

```

1 use std;
2 let double = fn(x) {
3     mul(2, x)
4 };
5 let square = fn(x) {
6     mul(x, x) # or pow(x, 2)
7 };
8 let main = double(square(8));

```

Elaine

Tuples are written as comma-separated lists of expressions surrounded with (). Tuples have a fixed length and can have elements of different types.

```

1 let main = (9, "hello");

```

Elaine

Additionally, Elaine features **if** expressions. The language does not support recursion or any other looping construct. Figure 4.1 contains a program that uses the basic features of Elaine and prints whether the square of 4 is even or odd.

```

1 # The standard library contains basic functions for manipulation
2 # of integers, booleans and strings.
3 use std;
4
5 # Functions are created with `fn` and bound with `let`, just like
6 # other values. The last expression in a function is returned.
7 let square = fn(x: Int) Int {
8     mul(x, x)
9 };
10
11 let is_even = fn(x: Int) Bool {
12     eq(0, modulo(x, 2))
13 };
14
15 # Type annotations can be inferred:
16 let square_is_even = fn(x) {
17     let result = is_even(square(x));
18     if result { "even" } else { "odd" }
19 };
20
21 let give_answer = fn(f, s, x) {
22     let prefix = concat(concat(s, " "), show_int(x));
23     let text = concat(prefix, " is ");
24     let answer = f(x);
25     concat(text, answer)
26 };
27
28 let main = give_answer(square_is_even, "The square of", 4);

```

Elaine

Figure 4.1: A simple Elaine program. The result of this program is the string "The square of 4 is even".

### 4.3 Types

Elaine is strongly typed with Hindley-Milner style type inference. Let bindings, function arguments, and function return types can be given explicit types. By convention, we will write variables and modules in lowercase. Types and type constructors are capitalized.

The primitive types are `String`, `Bool`, and `Int` for strings, booleans, and integers respectively.

```
1 | let x: Int = 5;           # ok!
2 | let x: String = 5;      # type error!
3 |
4 | let triple = fn(x: Int) Int { mul(3, x) };
5 | let y = triple("Hello"); # type error!
```

Elaine

We also could have written the type of the function as the type for the let binding. The type for a function is written like a function definition, without parameter names and body.

```
1 | let triple: fn(Int) Int = fn(x) { mul(3, x) };
```

Elaine

Type parameters start with a lowercase letter. Like in Haskell, they do not need to be declared explicitly.

```
1 | let f = fn(x: a) (a, Int) {
2 |     (x, 5)
3 | };
4 | let y = f("hello");
5 | let z = f(5);
```

Elaine

### 4.4 Algebraic Data Types

Complex programs often require custom data types. That is what the `type` construct is for. It is analogous to Koka's `type`, Haskell's `data` or Rust's `enum` construct.

A type declaration consists of a list of constructors each with a list of parameters. These constructors can be used as functions. A type can have type parameters, which are declared with `[]` after the type name. It is not possible to put constraints on type parameters.

Data types can be deconstructed with the `match` construct. The `match` construct looks like Rust's `match` or Haskell's `case`, but is more limited. It can only be used for custom data types and only matches on the outer constructor. For example, it is not possible to match on `Just(5)`, but only on `Just(x)`. Since the `Maybe` type is very common, it is provided in the standard library is the `maybe` module.

```

1 use std;
2
3 type Maybe[a] {
4     Just(a),
5     Nothing(),
6 }
7
8 let safe_div = fn(x, y) Maybe[Int] {
9     if eq(y, 0) {
10         Nothing
11     } else {
12         Just(div(x, y))
13     }
14 };
15
16 let main = match safe_div(5, 0) {
17     Just(x) => show_int(x),
18     Nothing => "Division by zero!",
19 };

```

Data types can be recursive. For example, we can define a `List` with a `Cons` and a `Nil` constructor.

```

1 type List[a] {
2     Cons(a, List[a]),
3     Nil(),
4 }
5
6 let list: List[Int] = Cons(1, Cons(2, Nil()));

```

The `List` type is also provided in the standard library in the `list` module. There is also syntactic sugar for lists: we can write a list with brackets and comma-separated expressions like `[1, 2, 3]`.

## 4.5 Recursion & Loops

The let-bindings in the previous sections are not allowed to be recursive. In general, let-bindings can only reference values that have been defined before the binding itself. However, recursion or some other looping construct is necessary for many programs. Therefore, Elaine has special syntax for recursive definitions: **let rec**. Let-bindings with **rec** are desugared into the Y combinator. However, it is impossible to write the Y combinator manually, because it would have an infinite type. The type checker therefore has special case for recursive definitions. An example of a recursive function is the `factorial` function listed below.



```

1 use std;
2
3 let rec factorial = fn(x: Int) {
4   if eq(x, 0) {
5     1
6   } else {
7     mul(x, factorial(sub(x,1)))
8   }
9 };

```

A word of caution: Elaine has no guards against unbounded recursion of functions or even recursive expressions. For example, the statements below are valid according to the Elaine type checker, but will cause infinite recursion when evaluated, which in practice means that it will run until the interpreter runs out of memory and crashes.

```

1 # Warning: these declarations will not halt!
2 let rec f = fn(x) { f(x) };
3 let rec x = x;

```

Using recursive definitions, we can define functions like `map`, `foldl`, and `foldr` to operate on our previously defined `List` type. The implementation for `map` might look like the listing below. Note that, in contrast with Haskell, Elaine evaluates these functions eagerly; there is no lazy evaluation.

```

1 let rec map = fn(f: fn(a) b, list: List[a]) List[b] {
2   match list {
3     Cons(a, as) => Cons(f(a), map(f, list)),
4     Nil() => Nil(),
5   }
6 };
7
8 let doubled = map(fn(x) { mul(2, x) }, [1, 2, 3]); # -> [2, 4, 6]

```

The `list` module provides the most common operations on lists. Such `head`, `concat_list`, `range`, `map`, `foldl`, and `foldr`. It also provides a `sum` function for lists of integers and a `join` function for lists of strings.

## 4.6 Modules

Modules can be used to group declarations. We have previously introduced the `std` module, which is available by default. However, we can also define new modules. The contents of any module can be brought into scope with the `use` declaration. However, only the public declarations in a module are imported. A declaration is public if it is prefixed with the `pub` keywords. This also applies to `mod` and `use` declarations. A public module can be imported after importing its parent module and a `pub use` brings the public declarations into the current module and makes them public.

```

1 use std;
2 mod math {
3     pub mod multiplication {
4         pub let square = fn(x: Int) Int { mul(x, x) };
5     }
6     pub let add3 = fn(x: Int, y: Int, z: Int) Int {
7         add(add(x, y), z)
8     };
9 }
10
11 use math;
12 use multiplication;
13
14 let main = add3(square(4), square(5), square(6));

```

## 4.7 Algebraic Effects

The programs in the previous sections are all pure and contain no effects. While a monadic approach is possible, Elaine provides first class support for algebraic effects and effect handlers to make working with effects more ergonomic. The design of effects in Elaine is heavily inspired by Koka (Leijen 2014).

An effect is declared with the **effect** keyword. An effect needs a name and a set of operations. Operations are the functions that are associated with the effect. They can have an arbitrary number of arguments and a return type. Only the signature of operations can be given in an effect declaration, the implementation must be provided via handlers (see Section 4.7.1).

Figure 4.2 lists examples of effect declarations for the **Abort**, **Ask**, **State**, and **Write** effects. We will refer to those declarations throughout this section. For the listings in this section, one can assume that these declarations are used. The **Abort** effect is meant to exit the computation. **Ask** provides some integer value to the computation, much like a global constant. **State** corresponds to the **State** monad in Haskell. Finally, **Write** allows us to write some string value somewhere. We will be using this to provide a substitute for writing to standard output.

```

1 effect Abort {
2     abort() (),
3 }

```

```

1 effect Ask {
2     ask() Int,
3 }

```

```

1 effect State {
2     get() Int,
3     put(Int) (),
4 }

```

```

1 effect Write {
2     write(String) (),
3 }

```

Figure 4.2: Examples of algebraic effect declarations for some simple effects.

### 4.7.1 Effect Handlers

To define the implementation of an effect, we have to define a handler it. Handlers are first-class values in Elaine and can be created with the **handler** keyword. They can then be

applied to an expression with the **handle** keyword. When **handle** expressions are nested with handlers for the same effect, the innermost **handle** applies.

For example, if we want to use an effect to provide an implicit value, we can make an effect `Ask` and a corresponding handler, which `resumes` execution with some values. The `resume` function represents the continuation of the program after the operation. The simplest handler for `Ask` we can write is one which yields some constant value.

```
1 | let hAsk = handler { ask() { resume(10) } };
2 |
3 | let main = handle[hAsk] add(ask(), ask()); # evaluates to 20
```

Elaine

Of course, it would be cumbersome to write a separate handler for every value we would like to provide. Since handlers are first-class values, we can return the handler from a function to simplify the code. This pattern is quite common to create dynamic handlers with small variations.

```
1 | let hAsk = fn(v: Int) {
2 |     handler { ask() { resume(v) } }
3 | };
4 |
5 | let main = {
6 |     let a = handle[hAsk(6)] add(ask(), ask());
7 |     let b = handle[hAsk(10)] add(ask(), ask());
8 |     add(a, b)
9 | };
```

Elaine

The true power of algebraic effects, however, lies in the fact that we can also write a handler with entirely different behaviour, without modifying the computation. For example, we can create a stateful handler which increments the value returned by `ask` on every call to provide unique identifiers. The program below will return 3, because the first `ask` call returns 1 and the second returns 2. This is accomplished in a very similar manner to the `State` monad.

```
1 | let hAsk = handler {
2 |     return(x) { fn(s: Int) { x } }
3 |     ask() {
4 |         fn(s: Int) {
5 |             let f = resume(s);
6 |             f(add(s, 1))
7 |         }
8 |     }
9 | };
10 |
11 | let c = handle[hAsk] add(ask(), ask());
12 | let main = c(1);
```

Elaine

Calling the `resume` function is not required. All effect operations are executed by the **handle** expression, hence, if we return from the operation, we return from the **handle** expression.

The `Abort` effect is an example which does not call the continuation. It defines a single operation `abort`, which stops the evaluation of the computation. The canonical handler for `Abort`, which returns the `Maybe` monad. If the computation returns, it should wrap the

returned value in `Just`. Otherwise, if the computation aborts, it should return `Nothing()`. In Elaine, if a sub-computation of a handler returns, the optional `return` arm of the handler will be applied. In the code below, this wraps the returned value in a `Just`. All arms of a handler must have the same return type.

```

1 effect Abort {
2   abort() a
3 }
4
5 let hAbort = handler {
6   return(x) { Just(x) }
7   abort() { Nothing() }
8 };
9
10 let main = handle[hAbort] {
11   abort();
12   5
13 };

```

Alternatively, we can define a handler that defines a default value for the computation in case it aborts. This is more convenient than the first handler if the `abort` case should always become

```

1 let hAbort = fn(default) {
2   handler {
3     return(x) { x }
4     abort() { default }
5   }
6 };
7
8 let safe_div = fn(x, y) <Abort> Int {
9   if eq(y, 0) {
10    abort()
11  } else {
12    div(x, y)
13  }
14 };
15
16 let main = add(
17   handle[hAbort(0)] safe_div(3, 0),
18   handle[hAbort(0)] safe_div(10, 2),
19 );

```

Just like we can ignore the continuation, we can also call it multiple times, which is useful for non-determinism and logic programming. In the listing below, the `Twice` effect is introduced, which calls its continuation twice. Combining that with the `State` effect as previously defined, the `put` operation is called twice, incrementing the initial state 3 by two, yielding a final result of 5. Admittedly, this example is a bit contrived. A more useful application of this technique can be found in Appendix A.1, which contains the full code for a very naive SAT solver in Elaine, using multiple continuations.

```

1 | effect Twice {
2 |     twice() ()
3 | }
4 |
5 | let hTwice = handler {
6 |     twice() {
7 |         resume();
8 |         resume()
9 |     }
10 | };
11 |
12 | let main = {
13 |     let a = handle[hState] handle[hTwice] {
14 |         twice();
15 |         put(add(get(), 1));
16 |         get()
17 |     };
18 |     a(3)
19 | };

```

### 4.7.2 Effect Rows

All types in Elaine have an effect row. So far, we have omitted the effect rows, because effect rows can be inferred by the type checker. Effect rows represent the set of effects that need to be handled to obtain the value in a computation. For simple values, that effect row is empty, denoted  $\langle \rangle$ . For example, an integer has type  $\langle \rangle$  `Int`. With explicit effect row, the square function in the previous section could therefore have been written as below.

```

1 | let square = fn(x:  $\langle \rangle$  Int)  $\langle \rangle$  Int {
2 |     mul(x, x)
3 | };

```

Simple effect rows consist of a list of effect names separated by commas. The return type of a function that returns an integer and uses the `Ask` and `State` effects has type  $\langle \text{Ask}, \text{State} \rangle$  `Int` or, equivalently  $\langle \text{State}, \text{Ask} \rangle$  `Int`. The order of effects in effect rows is irrelevant. However, the multiplicity is important, that is, the effect rows  $\langle \text{State}, \text{State} \rangle$  and  $\langle \text{State} \rangle$  are not equivalent.

Like in Koka, we can extend effect rows with other effect rows. This is denoted with the `|` at the end of the effect row:  $\langle A, B | e \rangle$  means that the effect row contains `A`, `B` and some (possibly empty) set of remaining effects. We call a row without extension *closed* and a row with extension *open*. Effect rows are considered equivalent up to reordering by the type checker. When a function is applied, the effect row of its return type is opened. This ensures that a function using only the state effect can be called in a computation wrapped in handlers for both the state and abort effect.

## 4.8 Functions Generic over Effects

We can use extensions of effect rows to ensure equivalence between effect rows without specifying the full rows. For example, the following function uses the `Abort` effect if the called function returns false, while retaining the effects of the wrapped function.

```

1 | let abort_on_false = fn(f: fn() <|e> Bool) <Abort|e> () {
2 |     if f() { () } else { abort() }
3 | };

```

Elaine

When an effect is handled, it is removed from the effect row. The `main` binding is required to have an empty effect row, which means that all effects in the program need to be handled. Therefore, to use the `abort_on_false` function defined above, it needs to be called from within a handler.

```

1 | let main: <> Maybe[()] = handle[hAbort] {
2 |     abort_on_false(fn() { false })
3 | };

```

Elaine

Recall the definition of `map` in Section 4.5, which was written without any effects in its signature. Adding the effect rows yields the following definition:

```

1 | let rec map = fn(f: fn(a) <|e> b, l: List[a]) <|e> List[b] {
2 |     match l {
3 |         Nil() => Nil(),
4 |         Cons(x, xs) => Cons(f(x), map(f, xs)),
5 |     }
6 | };

```

Elaine

Note that the parameter `f` and `map` use the same effect row variable `e`. This means that `map` has the same effect row as `f` for any effect row that `f` might have, including the empty effect row. This makes `map` quite powerful, because it can be applied in many situations.

```

1 | let pure_doubled = map(fn(x) { mul(2, x) }, [1,2,3]);
2 | let ask_added = handle[hAsk(5)] map(fn(x) { add(ask() x) }, [1,2,3]);

```

Elaine

If we were to write the same expressions in Haskell instead, we would need two different implementations of `map`: one for applying pure functions (`map`) and another for applying monadic functions (`mapM`). Our definition of `map` is therefore more general than Haskell's `map` function. The same reasoning can be applied to other functions like `foldl` and `foldr` or indeed any higher-order function.

Functional languages like Haskell usually do not feature constructs for looping, such as `while` loops. This is partly because folds, maps, and recursion are preferred to loops, but also because a looping construct relies on effects, since a loop does not return a value. In Elaine, we can define a `while` function which is generic over effects. This enables both functional and imperative styles of programming.

```

1  let rec while = fn(
2      predicate: fn() <|e> Bool,
3      body: fn() <|e> ()
4  ) <|e> () {
5      if predicate() {
6          body();
7          while(predicate, body)
8      } else {
9          ()
10     }
11 };

```

Elaine

## 4.9 Higher-Order Effects

Higher-order effects in Elaine are supported via elaborations, as introduced by Bach Poulsen and van der Rest (2023) and explained in Section 3.4. In this framework, higher-order effects are elaborated into a computation using only algebraic effects. They are not handled directly. This means that we cannot write handlers for them as we did for algebraic effects in the previous section.

To distinguish higher-order effects and operations from algebraic effects and operations, we write them with an `!` suffix. For example, a higher-order `Except!` effect is written `Except!`, and its `catch` operation is written `catch!`.

Higher-order effects are treated exactly like algebraic effects in the effect rows. The order of effects still does not matter, and we can create effect rows with arbitrary combinations of algebraic and higher-order effects.

The higher-order effect operations differ from other functions and algebraic operations because they have call-by-name semantics; the arguments are not evaluated before they are passed to the elaboration. Hence, the arguments can be effectful computations.

Just like we have the `handler` and `handle` keywords to create and apply handlers for algebraic effects, we can create and apply elaborations with the `elaboration` and `elab` keywords. Unlike handlers, elaborations do not get access to the `resume` function, and always resume exactly once.

An illustrative example of this feature is the `Reader` effect with a `local` operation. This effect enhances the previously introduced `Ask` effect with a `local` operation that modifies the value returned by `ask`. To motivate the implementation, let us first imagine how to emulate the behaviour of `local`. Our goal is to make the following snippet return the value `15`.

```

1  let main = handle[hAsk(5)] {
2      let x = ask();
3      let y = local(double, fn() { ask() });
4      add(x, y)
5  };

```

Elaine

This means that the `local` operation would need to handle the `ask` effect with the modified value. This is easily achieved, since the innermost handler always applies. If the function to modify the value is called `f`, then the value we should provide to the handler is `f(ask())`.

```

1  let local = fn(f: fn(Int) Int, g: fn() <Ask|e> a) <|e> a {
2      handle[hAsk(f(ask()))] { g() }
3  };

```

Elaine

This works but is not implemented as an effect. For example, we cannot modularly provide another implementation of `local`. To turn this implementation into an effect, we start with the effect declaration.

```
1 effect Reader! {
2     local!(fn(Int) Int, a) a
3 }
```

It might be surprising that the signature of `local` does not match the signature of the function above. That is because of the call-by-name nature of higher-order operations: instead of a function returning `a`, we simply have a computation that will evaluate to `a`. The effect row is irrelevant and therefore implicit. Now we can provide an elaboration, which is not a function, but better described as a syntactic substitution.

```
1 let eLocal = elaboration Reader! -> <Ask> {
2     local!(f, c) {
3         handle[hAsk(f(ask()))] c
4     }
5 };
```

Note how similar the elaboration for `local!` is to the `local` function above. In the first line, we specify explicitly what effect the elaboration elaborates (`Reader!`) and which effects should be present in the context where this elaboration is used (`<Ask>`). This can be an effect row of multiple effects if necessary. In this case we only require the `Ask` effect. This means that we can use this elaboration in any expression that is wrapped by at least a handler for `Ask`.

```
1 let main = handle[hAsk(5)] elab[eLocal] {
2     let x = ask();
3     let y = local!(double, ask());
4     add(x, y)
5 };
```

That is the full implementation for the higher-order `Reader!` effect in Elaine. Appendix A.2 contains a listing of all these pieces put together in a single example.

Another example is the `Except!` effect. This effect should allow us to use the `catch!` operation to recover from a `throw`. The latter is an algebraic, so we can start there.

```
1 type Result[a, b] {
2     Ok(a),
3     Err(b),
4 }
5
6 effect Throw {
7     throw(String) a
8 }
9
10 let hThrow = handler {
11     return(x) { Ok(x) }
12     throw(s) { Err(s) }
13 };
```



We assume here that we want to throw some string with an error message, but we could put a different type in there as well. The `throw` operation has a return type `a`, which is impossible to construct in general, so it cannot return. The higher-order `Except!` effect should then look like this:

```
1 effect Except! {
2     throw!(String) a
3     catch!(a, a) a
4 }
```

Elaine

In contrast with the `Reader!` effect above, we alias the operation of the underlying algebraic effect here. This makes no functional difference, except that it allows us to write functions with explicit effect rows with `Except!` and without `Throw`. We might even choose to elaborate to a different effect than `Throw`. The downside is that it requires us to provide the elaboration for the `throw!` operation.

```
1 let eExcept = elaboration Except! -> <Throw> {
2     throw!(s) { throw(s) }
3     catch!(a, b) {
4         match handle[hThrow] a {
5             Ok(x) => x,
6             Err(s) => b,
7         }
8     }
9 };
```

Elaine

We can then use the `Except!` effect like we used the `Reader!` effect: with an `elab` for `Except!` and a `handle` for `Throw`. In the listing below, we ensure that we do not decrement a value of `0` to ensure it will not become negative.

```
1 let main = handle[hThrow] elab[eExcept] {
2     let x = 0;
3     catch!({
4         if eq(x, 0) {
5             throw!("Whoa, x can't be zero!")
6         } else {
7             sub(x, 1)
8         }
9     }, 0)
10 };
```

Elaine

Since the elaborations can be swapped out, we can also design elaborations with different behaviour. Assume, for instance, that there is a `Log` effect. Then we can create an alternative elaboration that logs the errors it catches, which might be useful for debugging.

```

1 let eExceptLog = elaboration Except! -> <Throw,Log> {
2   throw!(s) { throw(s) }
3   catch!(a, b) {
4     match handle[hThrow] a {
5       Ok(x) => x,
6       Err(s) => {
7         log(s);
8         b
9       }
10    }
11  }
12 };

```

Elaine

We could also disable exception catching entirely if we so desire. This might be helpful if we are debugging a piece of a program that is wrapped in a `catch!` to ensure it never fully crashes, but we want to see errors while we are debugging. Of course, this changes the functionality of the program significantly. We should therefore be careful not to change computations that rely on a specific implementation of the `Except!`.

```

1 let eExceptIgnoreCatch = elaboration Except! -> <Throw> {
2   throw!(s) { throw(s) }
3   catch!(a, b) { a }
4 };

```

Elaine

What these examples illustrate is that elaborations provide a great deal of flexibility, with which we can define and alter the functionality of the `Except!` effect. We can change it temporarily for debugging purposes or apply another elaboration to a part of a computation. We can also define more effects like `Except!` and use them at the same time.



## Chapter 5

# Implicit Elaboration Resolution

With Elaine, we aim to explore further ergonomic improvements for programming with effects. We note that elaborations are often not parametrized and that there is often only one elaboration in scope per effect at a time. Hence, when we encounter an **elab**, there is only one possible elaboration that could be applied. Therefore, the language can infer the elaborations. Take the example in the listing below, where we let Elaine infer the elaboration.

```
1 let eLocal = elaboration Reader! -> <Ask> {
2     local!(f, c) {
3         handle[hAsk(f(ask()))] c
4     }
5 };
6
7 let main = handle[hAsk(2)] elab {
8     local!(double, add(ask(), ask()));
9 };
```

A use case of this feature is when an effect and elaboration are defined in the same module. When this module is imported, the effect and elaboration are both brought into scope and **elab** will apply the standard elaboration automatically.

```
1 mod local {
2     pub effect Ask { ... }
3     pub let hAsk = handler { ... }
4     pub effect Reader! { ... }
5     pub let eLocal = elaboration Reader! -> <Ask> { ... }
6 }
7
8 use local;
9
10 # We do not have to specify the elaboration, since it is
11 # imported along with the effect.
12 let main = handle[hAsk] elab { local!(double, ask!()) };
```

However, while useful, this feature only saves a few characters in the examples above. It becomes more important when multiple higher-order effects are involved: an **elab** without argument will elaborate all higher-order effects in the sub-computation. For instance, if elaborations for both **Except!** and **Reader!** are in scope, the following program works.

```

1 let main = handle[hAsk(2)] handle[hThrow] elab {
2     local!(double, {
3         if gt(ask(), 3) {
4             throw()
5         } else {
6             add(ask(), 4)
7         }
8     })
9 }

```

Elaine

To make the inference predictable, we require that an implicit elaboration must elaborate all higher-order effects in the sub-computation.

A problem with this feature arises when multiple elaborations for a single effect are in scope; which one should then be used? To keep the result of the inference predictable and deterministic, the type checker should yield a type error in this case. Hence, if type checking succeeds, then the inference procedure has found exactly one elaboration to apply for each higher-order effect. If not, the elaboration cannot be inferred and must be written explicitly.

```

1 let eLocal1 = elaboration Local! -> <Ask> { ... };
2 let eLocal2 = elaboration Local! -> <Ask> { ... };
3
4 let main = elab { local!(double, ask!()) }; # Type error here!

```

Elaine

The elaboration resolution consists of two parts: inference and transformation. The inference is done by the type checker and is hence type-directed, which records the inferred elaboration. After type checking the program is then transformed such that all implicit elaborations have been replaced by explicit elaborations.

To infer the elaborations, the type checker first analyses the sub-expression. This will yield some computation type with an effect row containing both higher-order and algebraic effects:  $\langle H1!, \dots, HN!, A1, \dots, AM \rangle$ . It then checks the type environment to look for elaborations  $e1, \dots, eN$  which elaborate  $H1!, \dots, HN!$ , respectively. Only elaborations that are directly in scope are considered, so if an elaboration resides in another module, it needs to be imported first. For each higher-order effect, there must be exactly one elaboration.

The **elab** is finally transformed into one explicit **elab** per higher-order effect. Recall that the order of elaborations does not matter for the semantics of the program, meaning that we apply them in arbitrary order.

A nice property of this transformation is that it results in readable code. Because the elaboration is in scope, there is an identifier for it in scope as well. The transformation simply inserts this identifier. The **elab** in the first example of this chapter will, for instance, be transformed to **elab**[eVal]. A code editor could then display this transformed **elab** as an inlay hint.

# Chapter 6

---

## Elaine Specification

This chapter contains the specification for Elaine. This specification includes the syntax definition, type inference rules, reduction semantics, and the functions provided by the standard library.

### 6.1 Syntax Definition

The Elaine syntax was designed to be easy to parse. The grammar is not white-space sensitive and most constructs are unambiguously identified with keywords at the start.

The full syntax definition is given in Figure 6.1 as a context-free grammar. In this grammar,  $x$  represents an identifier. We use the following notation for the grammar:

- a sort is declared with `::=`,
- the alternatives of a sort are separated by `|`,
- tokens are written in `monospace font`,
- $p^?$  indicates that the sort  $p$  is optional,
- $p \dots p$  indicates that the sort  $p$  can be repeated zero or more times, and
- $p, \dots, p$  indicates that the sort  $p$  can be repeated zero or more times, separated by commas.

The sorts  $x$ ,  $i$  and  $s$  denote identifiers, integers and strings respectively. Identifiers must match the regex expression `[a-zA-Z_][a-zA-Z_0-9]*` and cannot be a keyword. Integers consist of numeric digits. Strings are delimited by double quotes.

### 6.2 Effect Rows

The syntax definition in Figure 6.1 describes a layer of syntactic sugar for effect rows. The desugared syntax for effect rows is given as follows:

$$\Delta ::= \langle \rangle \mid \langle \phi \mid \Delta \rangle$$

We transform the syntactic sugar form into the standard form with the  $N$  operation defined in Figure 6.2a, which maps  $R$  to  $\Delta$ . This operation is applied to all occurrences of  $R$  in the syntax tree. Additionally, our typing judgments require operations that split an effect row into algebraic and higher-order effects, denoted  $A$  and  $H$ , respectively. The definition of these operations is given in Figure 6.2b.

Effect rows should be considered equivalent up to reordering. Following Leijen (2014), we define the  $\equiv$  relation to denote effect row equivalence in Figure 6.2c. This equivalence is implicitly used whenever effect rows are asserted to be equivalent in the typing judgments.

```

program  $p ::= d \dots d$ 

declaration  $d ::= \text{pub}^? \text{ mod } x \{d \dots d\}$ 
  |  $\text{pub}^? \text{ use } x;$ 
  |  $\text{pub}^? \text{ let rec}^? p = e;$ 
  |  $\text{pub}^? \text{ effect } \phi \{s, \dots, s\}$ 
  |  $\text{pub}^? \text{ type } x \{c, \dots, c\}$ 

block  $b ::= \{ es \}$ 
expression list  $es ::= e$ 
  |  $e; es$ 
  |  $\text{let rec}^? p = e; es$ 
expression  $e ::= x$ 
  |  $() \mid \text{true} \mid \text{false} \mid i \mid s$ 
  |  $(e, \dots, e)$ 
  |  $\text{fn}(p, \dots, p) T^? b$ 
  |  $\text{if } e \text{ b else } b$ 
  |  $e(e, \dots, e) \mid \phi(e, \dots, e)$ 
  |  $\text{handler } \{o, \dots, o\}$ 
  |  $\text{handler } \{\text{return}(x) b, o, \dots, o\}$ 
  |  $\text{handle}[e] e$ 
  |  $\text{elaboration } x! \rightarrow R \{o, \dots, o\}$ 
  |  $\text{elab}[e] e \mid \text{elab } e$ 
  |  $b$ 

annotatable variable  $p ::= x : T \mid x$ 
signature  $s ::= x(T, \dots, T) T$ 
effect clause  $o ::= x(x, \dots, x) b$ 
constructor  $c ::= x(T, \dots, T)$ 

type  $T ::= R \tau \mid \tau$ 
value type  $\tau ::= x$ 
  |  $() \mid \text{Bool} \mid \text{Int} \mid \text{String}$ 
  |  $\text{fn}(T, \dots, T) T$ 
  |  $\text{Handler}[x, \tau, \tau]$ 
  |  $\text{Elab}[x!, R]$ 
  |  $x[\tau, \dots, \tau]$ 
effect row sugar  $R ::= \langle \phi, \dots, \phi \rangle \mid \langle \phi, \dots, \phi \mid R \rangle \mid \Delta$ 
effect row  $\Delta ::= \langle \rangle \mid \langle \phi \mid \Delta \rangle$ 
effect  $\phi ::= x \mid x!$ 

```

Figure 6.1: Syntax definition of Elaine

$$\begin{aligned}
N(\langle \rangle) &= \langle \rangle \\
N(\langle | R \rangle) &= N(R) \\
N(\langle \phi_1 \rangle) &= \langle \phi_1 | \langle \rangle \rangle \\
N(\langle \phi_1, \dots, \phi_n \rangle) &= \langle \phi_1 | N(\langle \phi_2, \dots, \phi_n \rangle) \rangle \\
N(\langle \phi_1, \dots, \phi_n | R \rangle) &= \langle \phi_1 | N(\langle \phi_2, \dots, \phi_n | R \rangle) \rangle
\end{aligned}$$

(a) Definition of normalization  $N$  from the syntactic sugar for effect rows  $R$  to an effect row  $\Delta$ .

$$\begin{aligned}
H(\langle \rangle) &= \langle \rangle & A(\langle \rangle) &= \langle \rangle \\
H(\langle x | \Delta \rangle) &= H(\Delta) & A(\langle x | \Delta \rangle) &= \langle x | A(\Delta) \rangle \\
H(\langle x! | \Delta \rangle) &= \langle x! | H(\Delta) \rangle & A(\langle x! | \Delta \rangle) &= A(\Delta)
\end{aligned}$$

(b) Definition of the  $H$  and  $A$  operations, which reduce an effect row to only higher-order or algebraic effects, respectively.

$$\boxed{\Delta_1 \equiv \Delta_2}$$

$$\begin{array}{c}
\text{EQ-REFL} \\
\hline
\Delta \equiv \Delta
\end{array}
\qquad
\begin{array}{c}
\text{EQ-HEAD} \\
\Delta_1 \equiv \Delta_2 \\
\hline
\langle \phi | \Delta_1 \rangle \equiv \langle \phi | \Delta_2 \rangle
\end{array}$$
  

$$\begin{array}{c}
\text{EQ-TRANS} \\
\Delta_1 \equiv \Delta_2 \quad \Delta_2 \equiv \Delta_3 \\
\hline
\Delta_1 \equiv \Delta_3
\end{array}
\qquad
\begin{array}{c}
\text{EQ-SWAP} \\
\phi_1 \neq \phi_2 \\
\hline
\langle \phi_1 | \langle \phi_2 | \Delta \rangle \rangle \quad \langle \phi_2 | \langle \phi_1 | \Delta \rangle \rangle
\end{array}$$

(c) Effect row equivalence

Figure 6.2: Operations and relations on effect rows.

## 6.3 Type System

We give a declarative specification of the type system of Elaine. This specification consists of two parts: inference rules for expressions and inference rules for declarations.

The context  $\Gamma = (\Gamma_M; \Gamma_V; \Gamma_E; \Gamma_\Phi)$  consists of the following parts:

$\Gamma_M : x \rightarrow \Gamma$	module to context
$\Gamma_V : x \rightarrow \sigma$	variable to type scheme
$\Gamma_T : x \rightarrow T$	identifier to custom type
$\Gamma_\Phi : x \rightarrow \{s_1, \dots, s_n\}$	effect to operation signatures

In the typing judgments, we often need to extend just one of these sub-contexts. Therefore, if we extend one, the rest is implicitly passed to. For example, the following expressions are equivalent:

$$\begin{aligned}
\Gamma'_V &= \Gamma_V, x : T \\
\Gamma' &= (\Gamma_M ; \Gamma_V, x : T ; \Gamma_E ; \Gamma_\Phi)
\end{aligned}$$

In the typing judgments below, we assume that all types are explicitly specified for function arguments and return types. Hindley-Milner type inference can be used to infer the types



if missing. This is also done in the Elaine prototype. In addition to the typing judgments, we assert that all effects, effect operations and modules have unique names within any scope.

The typing judgments for expressions inductively define a ternary relation

$$\Gamma \vdash e : \Delta\tau,$$

where  $\Gamma$  ranges over contexts,  $e$  ranges over expressions and  $\Delta\tau$  ranges over pair of effect rows and value types. This relation should be read as: “in the context  $\Gamma$ , the expression  $e$  has type  $\Delta\tau$ ”. This relation is fairly standard, apart from the inclusion of the effect row. Note that in this specification, the effect row is an overapproximation of the effect which are present. This means, for example, that the (pure) boolean value **true** has the type  $\Delta\text{Bool}$  for every effect row  $\Delta$ , as rule E-TRUE shows. A common pattern in the judgments is that many expressions that are evaluated in sequence all have the same effect row. This can be seen clearly in the E-IF and E-TUPLE rules. A function application (E-APP) has an effect row that matches the same effect row on the return type of the called function.

Handlers have the type  $\text{Handler}[\phi, \tau, \tau']$ . In this type, the  $\phi$  variable represents the effect that it handles,  $\tau$  is the type of the sub-computation and  $\tau'$  is the type of the **handle** that handles a computation with type  $\tau$ . The  $\tau$  and  $\tau'$  types may be related. For example, a handler for the abort effect can be represented by the following type scheme:

$$\forall\alpha. \text{Handler}[\text{Abort}, \alpha, \text{Maybe}[\alpha]].$$

If no return branch is specified, it is assumed to be the identity function and hence  $\tau$  is equal to  $\tau'$  in that case, as written in rule E-HANDLER.

Elaborations have a similar type to handlers:  $\text{Elab}[\phi, \Delta]$ . Here,  $\phi$  again represents the effect this elaboration is for. The effect row  $\Delta$  is the set of algebraic effects it elaborates into. Therefore, E-ELABORATION asserts that  $\Delta$  is algebraic. When an elaboration is applied (E-ELAB), this effect row must be unifiable with the effect row for the return type of of **elab**.

Rule E-IMPLICITELAB is different from E-ELAB, because it elaborates all higher-order effects. Therefore, it requires that exactly one elaboration is in scope for each higher-order effect. In this rule,  $A(\Delta')$  represents the algebraic subset of the effects in  $\Delta'$ . The uniqueness is important here to keep the implicit elaboration resolution predictable as explained in Chapter 5.

The inference rules for declarations define a different relation

$$\Gamma \vdash d \Rightarrow (\Gamma_{\text{priv}}, \Gamma_{\text{pub}}),$$

where  $\Gamma$  again ranges over contexts and  $d$  ranges over declarations and sequences of declarations. The pair of contexts represent the private and public context that  $d$  generates. The relation should be read as “in the context  $\Gamma$ , the declaration  $d$  generates the bindings in  $\Gamma_{\text{priv}}$  and exposes the bindings in  $\Gamma_{\text{pub}}$ .” Both contexts only contain new bindings; they are not additive. The private context is always a subset of the public context.

All declarations generate only private bindings with an empty context for public bindings, written  $\varepsilon$ . The D-PUB rule then ensures that this private context is duplicated to the public context. Therefore, if the declarations in a module generate  $(\Gamma_{\text{priv}}; \Gamma_{\text{pub}})$ , then only  $\Gamma_{\text{pub}}$  should be stored in  $\Gamma_M$ , which is specified by the D-MOD rule.

The D-TYPE rule ensures two things: it adds the type to the type context  $\Gamma_T$  and adds all constructors as functions to the variable context  $\Gamma_V$ . Because all constructors are modelled as functions, we do not need special rules for constructors in the expression inference rules. The D-ALGEBRAICEFFECT and D-HIGHERORDEREFFECT rules are split to ensure that all higher-order operations have an ! suffix. Apart from that difference, these rules are identical.

$\Gamma \vdash e : T$			
$\frac{\text{E-GEN} \quad \Gamma \vdash e : \sigma \quad \alpha \notin \text{ftv}(\Gamma)}{\Gamma \vdash e : \forall \alpha. \sigma}$	$\frac{\text{E-INST} \quad \Gamma \vdash e : \forall \alpha. \sigma}{\Gamma \vdash e : \sigma[\alpha \mapsto T']}$	$\frac{\text{E-VAR} \quad \Gamma_V(x) = \Delta \tau}{\Gamma \vdash x : \Delta \tau}$	$\frac{\text{E-BLOCK} \quad \Gamma \vdash es : \Delta \tau}{\Gamma \vdash \{es\} : \Delta \tau}$
$\frac{\text{E-UNIT}}{\Gamma \vdash () : \Delta ()}$	$\frac{\text{E-INT}}{\Gamma \vdash i : \Delta \text{Int}}$	$\frac{\text{E-TRUE}}{\Gamma \vdash \text{true} : \Delta \text{Bool}}$	$\frac{\text{E-FALSE}}{\Gamma \vdash \text{false} : \Delta \text{Bool}}$
$\frac{\text{E-STRING}}{\Gamma \vdash s : \Delta \text{String}}$	$\frac{\text{E-TUPLE} \quad [\Gamma \vdash e_i : \Delta \tau_i]_{1 \leq i \leq n}}{\Gamma \vdash (e_1, \dots, e_n) : \Delta (\tau_1, \dots, \tau_n)}$	$\frac{\text{E-SEQ} \quad \Gamma \vdash e : \Delta \tau \quad \Gamma \vdash es : \Delta \tau'}{\Gamma \vdash e; es : \Delta \tau'}$	
$\frac{\text{E-LET} \quad \Gamma \vdash e : \Delta \tau \quad \Gamma_V, x : \tau \vdash es : \Delta \tau'}{\Gamma \vdash \text{let } x = e; es : \Delta \tau'}$	$\frac{\text{E-LETREC} \quad \Gamma_V, x : \tau \vdash e : \Delta \tau_1 \quad \Gamma_V, x : \tau \vdash es : \Delta \tau_2}{\Gamma \vdash \text{let rec } x = e; es : \Delta \tau_2}$		
$\frac{\text{E-FUNCDEF} \quad \Gamma_V, x_1 : \langle \rangle \tau_1, \dots, x_n : \langle \rangle \tau_n \vdash b : T}{\Gamma \vdash \text{fn}(x_1 : \tau_1, \dots, x_n : \tau_n) T b : \Delta (\tau_1, \dots, \tau_n) \rightarrow T}$		$\frac{\text{E-APP} \quad \Gamma \vdash e : (\tau_1, \dots, \tau_n) \rightarrow \Delta \tau \quad [\Gamma \vdash e_i : \Delta \tau_i]_{1 \leq i \leq n}}{\Gamma \vdash e(e_1, \dots, e_n) : \Delta \tau}$	
$\frac{\text{E-IF} \quad \Gamma \vdash e : \Delta \text{Bool} \quad \Gamma \vdash b_1 : \Delta \tau \quad \Gamma \vdash b_2 : \Delta \tau}{\Gamma \vdash \text{if } e \text{ } b_1 \text{ else } b_2 : \Delta \tau}$	$\frac{\text{E-MATCH} \quad \Gamma \vdash e : \Delta \tau \quad \{c_1, \dots, c_n\} = \Gamma_T(\tau) \quad \left[ \begin{array}{l} x_i(x_{i,1}, \dots, x_{i,m_i}) = p_i \quad x_i(\tau_{i,1}, \dots, \tau_{i,m_i}) = c_i \\ \Gamma, x_{i,1} : \tau_{i,1}, \dots, x_{i,m_i} : \tau_{i,m_i} \vdash e_i : \Delta \tau' \end{array} \right]_{1 \leq i \leq n}}{\Gamma \vdash \text{match } e \{ p_1 \Rightarrow e_1, \dots, p_n \Rightarrow e_n \} : \Delta \tau'}$		
$\frac{\text{E-ELAB} \quad \Gamma \vdash e_E : \Delta \text{Elab}[x!, \Delta] \quad \Gamma \vdash e_c : \langle x! \mid \Delta \rangle \tau}{\Gamma \vdash \text{elab}[e_E] e_c : \Delta \tau}$		$\frac{\text{E-IMPLICITELAB} \quad [\exists! x. \Gamma_V(x) = \text{Elab}[\phi, \Delta]]_{\phi \in \mathcal{H}(\Delta')} \quad \Gamma \vdash e : \Delta' \tau \quad \Delta = A(\Delta')}{\Gamma \vdash \text{elab } e : \Delta \tau}$	
$\frac{\text{E-HANDLE} \quad \Gamma \vdash e_h : \Delta \text{Handler}[\phi, \tau, \tau'] \quad \Gamma \vdash e_c : \langle \phi \mid \Delta \rangle \tau}{\Gamma \vdash \text{handle}[e_h] e_c : \Delta \tau'}$	$\frac{\text{E-HANDLERNORET} \quad \Gamma \vdash \text{handler} \{ \text{return } (x) \{x\}, o_1, \dots, o_n \} : \text{Handler}[\phi, \tau, \tau']}{\Gamma \vdash \text{handler} \{ o_1, \dots, o_n \} : \text{Handler}[\phi, \tau, \tau']}$		
$\frac{\text{E-HANDLER} \quad \Gamma_{\Phi}(\phi) = \{s_1, \dots, s_n\} \quad \Gamma, x : \tau \vdash e_{\text{ret}} : \tau' \quad \left[ \begin{array}{l} s_i = x_i(\tau_{i,1}, \dots, \tau_{i,m_i}) \rightarrow \tau_i \quad o_i = x_i(x_{i,1}, \dots, x_{i,m_i}) \{e_i\} \\ \Gamma_V, \text{resume} : (\tau_i) \rightarrow \tau', x_{i,1} : \tau_{i,1}, \dots, x_{i,m_i} : \tau_{i,m_i} \vdash e_i : \tau' \end{array} \right]_{1 \leq i \leq n}}{\Gamma \vdash \text{handler} \{ \text{return } (x) \{e_{\text{ret}}\}, o_1, \dots, o_n \} : \text{Handler}[\phi, \tau, \tau']}$			
$\frac{\text{E-ELABORATION} \quad \Gamma_{\Phi}(x!) = \{s_1, \dots, s_n\} \quad \Delta = A(\Delta) \quad \left[ \begin{array}{l} s_i = x_i!(\tau_{i,1}, \dots, \tau_{i,m_i}) \tau_i \quad o_i = x_i!(x_{i,1}, \dots, x_{i,m_i}) \{e_i\} \\ \Gamma, x_{i,1} : \Delta \tau_{i,1}, \dots, x_{i,m_i} : \Delta \tau_{i,m_i} \vdash e_i : \Delta \tau_i \end{array} \right]_{1 \leq i \leq n}}{\Gamma \vdash \text{elaboration } x! \rightarrow \Delta \{o_1, \dots, o_n\} : \text{Elab}[x!, \Delta]}$			

Figure 6.3: Inference rules for expressions.

## 6.4 Desugaring

To simplify the reduction rules, we simplify the AST by desugaring some constructs. This transform is given by a fold over the syntax tree with the operation  $D$  defined below. If a construct does not match one of the cases below, it is not transformed.

$$\begin{aligned}
D(\text{fn}(x_1 : T_1, \dots, x_n : T_n) T \{e\}) &= \lambda x_1, \dots, x_n. e \\
D(\text{let } x = e_1; e_2) &= (\lambda x. e_2)(e_1) \\
D(\text{let rec } x = e_1; e_2) &= (\lambda x. e_2)((\lambda x. x(x))(\lambda x. e_1[x \mapsto x(x)])) \\
D(e_1; e_2) &= (\lambda \_ . e_2)(e_1) \\
D(\{e\}) &= e
\end{aligned}$$

## 6.5 Semantics

We give two descriptions of the semantics of Elaine. First, we give an informal description of the denotational semantics to highlight the connection between Elaine and hefty algebras. There is no full specification for the denotational semantics of Elaine. Second, we give a formal small-step reduction semantics for the evaluation of expressions.

### 6.5.1 Denotational Semantics

In this section, we give an intuition for the intended denotational semantics of Elaine. The  $\llbracket \cdot \rrbracket$  operator is used to represent the denotational semantics.

In languages based on algebraic effects, such as Koka, the semantics of effect rows are free monad. So, the denotational semantics of Koka's effect row notation would then be given as follows:

$$\llbracket () \rightarrow e \ a \rrbracket = () \rightarrow \text{Free } \llbracket e \rrbracket \ \llbracket a \rrbracket.$$

However, the semantics for Elaine's effect rows are hefty trees instead of free monads. Therefore, effects correspond to higher-order functors of a hefty tree and algebraic effects need to be lifted. In general, this can be expressed as:

$$\llbracket \text{fn}() \rightarrow e \ a \rrbracket = () \rightarrow \text{Hefty } (\text{Lift } \llbracket A(e) \rrbracket + \llbracket H(e) \rrbracket) \ \llbracket a \rrbracket.$$

We conjecture that the denotational semantics of Elaine expressions are hefty trees. However, the semantics of Elaine are still unclear, because the **elab** construct does not correspond directly to hefty algebras. To see why, recall that elaboration in hefty algebras returns a free monad. To elaborate multiple higher-order effects, the elaborations of the effects need to be composed using  $\wedge$  operator. Elaine's **elab** construct, however, returns a type with only the elaborated effect removed and allows other higher-order effects to be left in the effect row. Composition of elaborations is then no longer necessary, because individual elaborations can be applied in sequence. For example, to elaborate a computation that uses both the catch and reader effect, Elaine allows the following expression:

```
1 | let main = elab[eCatch] elab[eReader] { ... };
```

Elaine

This provides more flexibility than composition, since elaborations can then be applied to specific sub-computations of elaborations. This means that Elaine is more flexible than hefty algebras. For example, the following example does not have a clear semantics as hefty tree:

```

1 | # Note: valid in Elaine, but not in hefty algebras!
2 | let main = elab[eReader] {
3 |     elab[eCatch1] a();
4 |     elab[eCatch2] b()
5 | };

```

Elaine

Therefore, the semantics for Elaine’s **elab** construct are left to future work.

### 6.5.2 Small-step Reduction Semantics

The reduction semantics for Elaine are given in Figure 6.5. It is given in the form of two contexts  $E$  and  $X_{op}$  and a reduction relation  $\longrightarrow$ .

The  $E$  context is used for all reduction rules except effect operations, such as **if**, **let**, and function applications. The  $X_{op}$  context is the context in which a handler can reduce an operation  $op$ . The two contexts are mostly equivalent, except for the fact that an  $X_{op}$  cannot enter the sub-computation of a **handle** or **elab** construct. This is important to ensure correct behaviour for when there are multiple nested handlers for a single effect. In that case, only the innermost handler should be able to handle said effect. The same reasoning applies to elaborations. We use the  $\in$  symbol to mean that an operation is not handled by a handler, or elaborated by an elaboration.

Figure 6.5 with semantics does not include semantics for declarations and modules. A sequence of declarations is evaluated in order. The bindings from each declaration get substituted in the remainder of the program. If the declarations are inside a module declaration, then the public bindings get collected. These bindings are substituted when a module is imported with the **use** declaration.

In this semantics, we assume that all elaborations are explicit. If they are implicit, they first need to be transformed according to the procedure from Chapter 5.

## 6.6 Standard Library

To simplify parsing, Elaine does not include any operators. For the lack of operators, any manipulation of primitives needs to be done via the standard library of built-in functions. These functions reside in the **std** module, which can be imported like any other module with the **use** statement to bring its contents into scope.

The full list of functions available in the **std** module, along with their signatures and descriptions, is given in Figure 6.6a. The **std** module also contains several other modules that can be imported. These modules are all written in Elaine itself. An overview of these modules is given in Figure 6.6b with the signatures of the functions and types they contain.

$$\boxed{\Gamma \vdash d \Rightarrow (\Gamma_{\text{priv}}; \Gamma_{\text{pub}})}$$

$$\begin{array}{c}
\text{D-SEQ} \\
\frac{\Gamma \vdash d_1 \Rightarrow (\Gamma_{\text{priv},1}, \Gamma_{\text{pub},1}) \quad \Gamma, \Gamma_{\text{priv},1} \vdash d_2 \dots d_n \Rightarrow (\Gamma_{\text{priv},n}, \Gamma_{\text{pub},n})}{\Gamma \vdash d_1 \dots d_n \Rightarrow (\Gamma_{\text{priv},1}, \Gamma_{\text{priv},n}; \Gamma_{\text{pub},1}, \Gamma_{\text{pub},n})}
\end{array}$$

$$\begin{array}{c}
\text{D-MOD} \\
\frac{\Gamma \vdash d_1 \dots d_n \Rightarrow (\Gamma_{\text{priv}}; \Gamma_{\text{pub}}) \quad \Gamma_{M,x} = x : \Gamma_{\text{pub}}}{\Gamma \vdash \text{mod } x \{d_1 \dots d_n\} \Rightarrow (\Gamma_{M,x}; \varepsilon)}
\end{array}$$

$$\begin{array}{c}
\text{D-PUB} \\
\frac{\Gamma \vdash d \Rightarrow (\Gamma'; \varepsilon)}{\Gamma \vdash \text{pub } d \Rightarrow (\Gamma'; \Gamma')}
\end{array}$$

$$\begin{array}{c}
\text{D-USE} \\
\frac{}{\Gamma \vdash \text{use } x; \Rightarrow (\Gamma_M(x); \varepsilon)}
\end{array}$$

$$\begin{array}{c}
\text{D-LET} \\
\frac{\Gamma \vdash e : \tau \quad \Gamma'_V = x : \tau}{\Gamma \vdash \text{let } x = e; \Rightarrow (\Gamma'; \varepsilon)}
\end{array}$$

$$\begin{array}{c}
\text{D-LETREC} \\
\frac{\Gamma, x : \tau \vdash e : \tau \quad \Gamma'_V = x : \tau}{\Gamma \vdash \text{let rec } x = e; \Rightarrow (\Gamma'; \varepsilon)}
\end{array}$$

$$\begin{array}{c}
\text{D-TYPE} \\
\frac{\Gamma'_V = x_1 : f_1, \dots, x_m : f_m \quad \Gamma'_T = x : \{x_1(\tau_{1,1}, \dots, \tau_{1,n_1}), \dots, x_m(\tau_{m,1}, \dots, \tau_{m,n_m})\} \quad f_i = \forall \alpha. (\tau_{i,1}, \dots, \tau_{i,n_i}) \rightarrow \alpha x}{\Gamma \vdash \text{type } x \{x_1(\tau_{1,1}, \dots, \tau_{1,n_1}), \dots, x_m(\tau_{m,1}, \dots, \tau_{m,n_m})\} \Rightarrow (\Gamma'; \varepsilon)}
\end{array}$$

$$\begin{array}{c}
\text{D-ALGEBRAICEFFECT} \\
\frac{s_i = \text{op}_i(\tau_{i,1}, \dots, \tau_{i,n_i}) : \tau_i \quad \Gamma'_\Phi(x) = \{s_1, \dots, s_n\}}{\Gamma \vdash \text{effect } x \{s_1, \dots, s_n\} : (\Gamma'; \varepsilon)}
\end{array}$$

$$\begin{array}{c}
\text{D-HIGHERORDEREFFECT} \\
\frac{s_i = \text{op}_i!(\tau_{i,1}, \dots, \tau_{i,n_i}) : \tau_i \quad \Gamma'_\Phi(x!) = \{s_1, \dots, s_n\}}{\Gamma \vdash \text{effect } x! \{s_1, \dots, s_n\} : (\Gamma'; \varepsilon)}
\end{array}$$

Figure 6.4: Inference rules for declarations.

$$\begin{aligned}
E ::= & [] \mid E(e_1, \dots, e_n) \mid v(v_1, \dots, v_n, E, e_1, \dots, e_m) \\
& \mid \text{if } E \{e_1\} \text{ else } \{e_2\} \\
& \mid \text{match } E \{ p_1 \Rightarrow e_1, \dots, p_n \Rightarrow e_n \} \\
& \mid \text{let } x = E; e \mid E; e \\
& \mid \text{handle}[E] e \mid \text{handle}[v] E \\
& \mid \text{elab}[E] e \mid \text{elab}[v] E \\
\\
X_{op} ::= & [] \mid X_{op}(e_1, \dots, e_n) \mid v(v_1, \dots, v_n, X_{op}, e_1, \dots, e_m) \\
& \mid \text{if } X_{op} \{e_1\} \text{ else } \{e_2\} \\
& \mid \text{match } X_{op} \{ p_1 \Rightarrow e_1, \dots, p_n \Rightarrow e_n \} \\
& \mid \text{let } x = X_{op}; e \mid X_{op}; e \\
& \mid \text{handle}[X_{op}] e \mid \text{handle}[h] X_{op} \text{ if } op \notin h \\
& \mid \text{elab}[X_{op}] e \mid \text{elab}[\epsilon] X_{op} \text{ if } op \notin \epsilon \\
\\
(\lambda x_1, \dots, x_n. e)(v_1, \dots, v_n) & \longrightarrow e[x_1 \mapsto v_1, \dots, x_n \mapsto v_n] \\
\text{if true } \{e_1\} \text{ else } \{e_2\} & \longrightarrow e_1 \\
\text{if false } \{e_1\} \text{ else } \{e_2\} & \longrightarrow e_2 \\
\text{match } c(v_1, \dots, v_n) \{ & \\
\quad \dots & \\
\quad c(x_1, \dots, x_n) \Rightarrow e & \longrightarrow e[x_1 \mapsto v_1, \dots, x_n \mapsto v_n] \\
\quad \dots & \\
\} & \\
\text{handle}[h] v & \longrightarrow e[x \mapsto v] \\
& \quad \text{where } \text{return}(x)\{e\} \in h \\
\text{handle}[h] X_{op}[op(v_1, \dots, v_n)] & \longrightarrow e[x_1 \mapsto v_1, \dots, x_n \mapsto v_n, \text{resume} \mapsto k] \\
& \quad \text{where } op(x_1, \dots, x_n)\{e\} \in h \\
& \quad \quad k = \lambda y. \text{handle}[h] X_{op}[y] \\
\text{elab}[\epsilon] v & \longrightarrow v \\
\text{elab}[\epsilon] X_{op}[op!(e_1, \dots, e_n)] & \longrightarrow \text{elab}[\epsilon] X_{op}[e[x_1 \mapsto e_1, \dots, x_n \mapsto e_n]] \\
& \quad \text{where } op!(x_1, \dots, x_n)\{e\} \in \epsilon
\end{aligned}$$

Figure 6.5: Reduction semantics for Elaine.

	Name	Type signature		Description
Arithmetic	add	<b>fn</b> (Int, Int)	Int	addition
	sub	<b>fn</b> (Int, Int)	Int	subtraction
	neg	<b>fn</b> (Int)	Int	negation
	mul	<b>fn</b> (Int, Int)	Int	multiplication
	div	<b>fn</b> (Int, Int)	Int	division
	modulo	<b>fn</b> (Int, Int)	Int	modulo
	pow	<b>fn</b> (Int, Int)	Int	exponentiation
Comparisons	eq	<b>fn</b> (Int, Int)	Bool	equality
	neq	<b>fn</b> (Int, Int)	Bool	inequality
	gt	<b>fn</b> (Int, Int)	Bool	greater than
	geq	<b>fn</b> (Int, Int)	Bool	greater than or equal
	lt	<b>fn</b> (Int, Int)	Bool	less than
	leq	<b>fn</b> (Int, Int)	Bool	less than or equal
Boolean operations	not	<b>fn</b> (Bool)	Bool	boolean negation
	and	<b>fn</b> (Bool, Bool)	Bool	boolean and
	or	<b>fn</b> (Bool, Bool)	Bool	boolean or
String operations	concat	<b>fn</b> (Bool, Bool)	Bool	string concatenation
	is_prefix	<b>fn</b> (String, String)	Bool	is prefix of
	str_eq	<b>fn</b> (String, String)	Bool	string equality
	drop	<b>fn</b> (Int, String)	String	drop characters
	take	<b>fn</b> (Int, String)	String	take characters
	length	<b>fn</b> (String)	Int	string length
Conversions	show_int	<b>fn</b> (Int)	String	integer to string
	show_bool	<b>fn</b> (Bool)	String	integer to string

(a) The built-in functions in the `std` module in Elaine.

Module	Item	Type signature
loop	while	<b>fn</b> ( <b>fn</b> () < e> Bool, <b>fn</b> () < e> ()) < e> ()
	repeat	<b>fn</b> (Int, <b>fn</b> (Int) < e> ()) < e> ()
maybe	Maybe[a]	<b>type</b> Maybe[a] { Just(a), Nothing() }
abort	abort	<b>effect</b> Abort { abort() a }
	hAbort	Handler Abort a Maybe[a]
list	List	<b>type</b> List[a] { Cons(a, List[a]), Nil() }
	head	<b>fn</b> (List[a]) Maybe[a]
	concat_list	<b>fn</b> (List[a], List[a]) List[a]
	range	<b>fn</b> (Int, Int) List[Int]
	map	<b>fn</b> ( <b>fn</b> (a) < e> b, List[a]) < e> List[b]
	foldl	<b>fn</b> ( <b>fn</b> (a, b) < e> b, b, List[a]) < e> List[b]
	foldr	<b>fn</b> ( <b>fn</b> (a, b) < e> b, b, List[a]) < e> List[b]
	sum	<b>fn</b> (List[Int]) Int
	join	<b>fn</b> (List[String]) String
	explode	<b>fn</b> (String) List[String]
state	State	<b>effect</b> State { get() Int, put(Int) () }
	hState	Handler[State, a, <b>fn</b> (Int) a]
state_str	State	<b>effect</b> State { get() String, put(String) () }
	hState	Handler[State, a, <b>fn</b> (String) a]

(b) Sub-modules of `std`

Figure 6.6: Overview of Elaine's standard library.

# Chapter 7

---

## Related Work

This chapter discusses extensions to algebraic effects and alternatives to algebraic effects and hefty algebras. Additionally, we discuss some other languages with algebraic effects.

### 7.1 Monad Transformers

Monad transformers provide a way to compose monads (Moggi 1989a). This makes them an alternative to the free monad. While monad transformers predate algebraic effects, they do support higher-order effects. A popular implementation of monad transformers is Haskell's `mtl`<sup>1</sup> library. In the rest of this section, we adopt the terminology from that library.

The goal of monad composition is to make the operations of all composed monads available to the computation. Given two monads `A` and `B`, a naive composition would result in the type `A (B a)`. However, this type represents a computation using `A` that returns a computation `B a`, meaning that it is not possible to use operations of both monads.

A monad transformer is a type constructor that takes some monad and returns a new monad. Usually, the transformation it performs is to add operations to the input monad. Composing `A` and `B` then requires some transformer `AT` to be defined, such that `AT B` is a monad that provides the operations of both `A` and `B`. An arbitrary number of monad transformers can be composed this way. The representation of a monad then becomes much like that of a list of monad transformers. The `Identity` monad marks the end of the list, and is defined as below.

```
1 | newtype Identity a = Identity a
```

Haskell

A neat property of monad transformers is that a monad can be easily obtained by applying the transformer to the identity monad. Haskell's `mtl` library, for instance, defines a monad transformer `StateT` and then defines `State` as `StateT Identity`. The operations of the state effect are then not implemented on `StateT` directly, but on are part of a type class `MonadState`. The `StateT` is then an instance of `MonadState` class. Every other transformer is an instance of `MonadState` if its input monad is an instance of `MonadState`. For example, for the `WriterT` instance, there is the following instance declaration.

```
1 | instance MonadState s (StateT s m) where  
2 |   -- definitions omitted  
3 |  
4 | instance MonadState s m => MonadState s (WriteT m) where  
5 |   -- definitions omitted
```

Haskell

---

<sup>1</sup><https://github.com/haskell/mtl>



A computation can then be generic over the monad transformers, requiring only that `StateT` is present somewhere in the stack of monad transformers.

```

1 | usesState :: MonadState Int m => Int -> m Int
2 | usesState a = get >>= \x -> put (x + a)

```

Haskell

This is analogous to the `State s < f` constraint from the free monad encoding. However, there is a cost to this approach. For every effect, a new type class needs to be introduced and there need to be instance definitions on all existing monad transformers. The number of instance declarations therefore scales quadratically with the number of effects.

Another downside to monad transformers is that the order in which the monads need to be evaluated is entirely fixed. In the free monad encoding and languages with algebraic effects, the effects in the effect row can be reordered. To evaluate a computation with monad transformers, the transformers need to be run one at a time. The order of the monad transformers determines the order in which they must be run: the outermost monad transformer must be run first. This is in contrast with algebraic effects, which can be handled in any order.

## 7.2 Other Solutions to the Modularity Problem

An alternative to hefty algebras for solving the modularity problem is the theory of *scoped effects* (Piróg et al. 2018; Wu, Schrijvers, and Hinze 2014; Yang et al. 2022). This theory replaces the free monad by a `Prog` monad, which features one additional constructor called `Enter`. Along with the continuation, this constructor takes a sub-computation. The return value of this sub-computation is passed to the continuation. In that sense, the `Enter` constructor matches the `>>=`, but without distributing the continuation over its sub-computation.

Instead of defining evaluation as a single algebra, scoped effects requires two algebras: an endo-algebra for scoped operations and a base-algebra for the other operations. This is somewhat similar to the distinction between elaboration and handling for hefty algebras, however, in hefty algebras, the algebras are not applied at the same time.

Many higher-order effects, such as the exception and reader effects, can be expressed in this framework. However, it is less general than hefty algebras, because there are some higher-order effects that cannot be expressed as scoped effects. This concerns effects that defer some computation, such as the lambda abstraction (van den Berg et al. 2021). Hefty algebras are therefore more general than scoped effects (Bach Poulsen and van der Rest 2023).

The limitations of scoped effects can be understood intuitively by emulating them in Elaine. The endo-algebra of scoped effects corresponds roughly with a `handle` operation in an elaboration. Since the result of the sub-computation in scoped effect must directly be passed to the continuation, the elaboration contains only a `handle` and nothing else. Therefore, any higher-order effect that can be expressed as the elaboration below (up to renaming) can be defined in the theory of scoped effects. However, this is an informal and imperfect comparison, since scoped effects and hefty algebras have very different semantics.

```

1 | effect ScopedEffect! {
2 |     scoped_operation!(a) a
3 | }
4 |
5 | let eScoped = elaboration ScopedEffect! -> AlgebraicEffect {
6 |     scoped_operation!(a) {
7 |         handle[endoAlg] a
8 |     }
9 | };

```

Scoped effects have been generalized by van den Berg et al. (2021) to *latent effects*, which supports the same set of effects as hefty algebras. Bach Poulsen and van der Rest (2023) note that while latent effects are powerful, they require *weaving glue* to ensure unhandled operations are treated correctly through sub-computations. In contrast, hefty algebras do not require any weaving.

### 7.3 Languages with Native Support for Algebraic Effects

The motivation of adding support for effects to a programming language is twofold. First, it enables effects to be implemented into languages with type systems in which effects cannot be encoded as a free monad or a similar model. Second, built-in effects allow for more ergonomic and performant implementations. Naturally, the ergonomics of any given implementation are subjective, but we undeniably have more control over the syntax by adding effects to the language.

Notable examples of languages with first-class support for algebraic effects are Koka (Leijen 2014), Frank (Lindley, McBride, and McLaughlin 2017), Effekt (Bach Poulsen and van der Rest 2023), Eff (Bauer and Pretnar 2015), Helium (Biernacki et al. 2019), and OCaml (Sivaramakrishnan et al. 2021). In all of these languages, effect row variables can be used to abstract over effects. For example, the signature of the `map` function in Koka is given below and is similar to the signature of `map` in Elaine.

```

1 | fun map ( xs : list<a>, f : a -> e b ) : e list<b>
2 |     ...

```

Other languages choose a more implicit syntax for effect polymorphism. Frank (Lindley, McBride, and McLaughlin 2017) opts to have the empty effect row represent the *ambient effects*. Any effect row then becomes not the exact set of effects that need to be handled, but the smallest set. The equivalent signature of `map` is then written as

```

1 | map : {X -> []Y} -> List X -> []List Y

```

In contrast with Elaine, languages such as Koka and Frank do not have dedicated types for handlers and `handle` constructs. Instead, they represent handlers as functions that take computations as arguments. In Elaine, there are dedicated types and constructs for effect handlers so that they are symmetric with elaborations. That is, the counterpart of `elab` is `handle` and the counterpart of `elaboration` is `handler`.

Koka implements several extensions to standard algebraic effects. First, it supports named handlers (Xie et al. 2022), which provide a mechanism to distinguish between multiple occurrences of an effect in an effect row. Additionally, Koka features *scoped handlers*, which are different from the previously mentioned scoped effects. Scoped handlers make it possible to associate types with handler instances (Xie et al. 2022).

## 7.4 Effects as Free Monads

There are many libraries that implement the free monad in various forms in Haskell, including `fused-effects`<sup>2</sup>, `polysemy`<sup>3</sup>, `freer-simple`<sup>4</sup>, and `eff`<sup>5</sup>. Each of these libraries give the encoding of effects a slightly different spin in an effort to find the most ergonomic and performant representation. They are all not just based on the free monad, but on freer monads (Kiselyov and Ishii 2016) and fused effects (Wu and Schrijvers 2015) which yield better performance than the free monad. Some of these libraries support scoped effects as well, but apart from the work by Bach Poulsen and van der Rest (2023), no libraries with support for hefty algebras have been published.

Effect rows are often constructed using the *Data Types à la Carte* technique (Swierstra 2008), which requires a fairly robust type system. Hence, many languages cannot encode effects within the language itself. In some languages, it is possible to work around the limitations with metaprogramming, such as the Rust library `effin-mad`<sup>6</sup>, though the result does not integrate well with the rest of language and its use in production is strongly discouraged by the author.

The programming language Idris (Brady 2013) also has an implementation of algebraic effects in its standard library. It is an interesting case study since Idris is a dependently typed language. Due to its dependent typing, it can distinguish multiple occurrences of a single effect in the same effect row by assigning them different *labels*. This is similar to what *named handlers* (Xie et al. 2022) aim to accomplish.

---

<sup>2</sup><https://github.com/fused-effects/fused-effects>

<sup>3</sup><https://github.com/polysemy-research/polysemy>

<sup>4</sup><https://github.com/lexi-lambda/freer-simple>

<sup>5</sup><https://github.com/hasura/eff>

<sup>6</sup><https://github.com/rosefromthedead/effing-mad>

# Chapter 8

---

## Conclusion

We have presented a language specification based on hefty algebras. This specification includes the syntax definition, typing judgements and reduction semantics. Our focus in this endeavour was to show the viability and explore the ergonomics of such a language. The result is, in our opinion, a language in which higher-order effects can be represented with relative ease.

Additionally, we introduced implicit elaboration resolution, which reduces the syntactic overhead of elaborations. Of particular interest is how this feature interacts with the module system for any language, to allow effects to be imported along with their elaborations.

The prototype implementing this specification allowed us to create a set of example programs available in Appendix A and in the artifact, which show off various applications of elaborations.

### 8.1 Future Work

Because Elaine’s **elab** does not require all elaboration to be applied at once, the denotational semantics of Elaine are unclear. We conjecture that the semantics of expressions in Elaine are hefty trees, but we have no proof for this claim. This is a gap that could be filled by future work.

A missing feature in Elaine is type parameters for effects. In Koka, for example, the state effect `state<s>` is parametrized by a type `s`, which is not possible in Elaine. Another omission are IO operations. An Elaine program cannot write to files, accept input or print text apart from the value it returns. Furthermore, Elaine does not include any extensions of algebraic effects, such as named handlers.

The prototype for Elaine only features an interpreter, not a compiler. So, another direction for future work is towards efficient compilation of elaborations. Since compilation of algebraic effects is well-established (Leijen 2017), a transformation from a program with elaborations to a program that only uses algebraic effects should enable full compilation of program with higher-order effects.



---

# Bibliography

- Bach Poulsen, Casper (2023a). *Algebras of Higher-Order Effects in Haskell*. URL: <http://casperbp.net/posts/2023-08-algebras-of-higher-order-effects/> (visited on 09/09/2023).
- (2023b). *Encoding Higher-Order Effects in Haskell*. URL: <http://casperbp.net/posts/2023-08-encoding-higher-order-effects/> (visited on 09/22/2023).
- Bach Poulsen, Casper and Cas van der Rest (Jan. 9, 2023). “Hefty Algebras: Modular Elaboration of Higher-Order Algebraic Effects”. In: *Proceedings of the ACM on Programming Languages* 7 (POPL), pp. 1801–1831. ISSN: 2475-1421. DOI: 10.1145/3571255. URL: <https://dl.acm.org/doi/10.1145/3571255> (visited on 01/26/2023).
- Bauer, Andrej and Matija Pretnar (Jan. 2015). “Programming with algebraic effects and handlers”. In: *Journal of Logical and Algebraic Methods in Programming* 84.1, pp. 108–123. ISSN: 23522208. DOI: 10.1016/j.jlamp.2014.02.001. URL: <https://linkinghub.elsevier.com/retrieve/pii/S2352220814000194> (visited on 04/03/2023).
- Van den Berg, Birthe et al. (2021). “Latent Effects for Reusable Language Components”. In: *Programming Languages and Systems*. Ed. by Hakjoo Oh. Vol. 13008. Series Title: Lecture Notes in Computer Science. Cham: Springer International Publishing, pp. 182–201. ISBN: 978-3-030-89050-6 978-3-030-89051-3. DOI: 10.1007/978-3-030-89051-3\_11. URL: [https://link.springer.com/10.1007/978-3-030-89051-3\\_11](https://link.springer.com/10.1007/978-3-030-89051-3_11) (visited on 01/12/2023).
- Biernacki, Dariusz et al. (Jan. 2, 2019). “Abstracting algebraic effects”. In: *Proceedings of the ACM on Programming Languages* 3 (POPL), pp. 1–28. ISSN: 2475-1421. DOI: 10.1145/3290319. URL: <https://dl.acm.org/doi/10.1145/3290319> (visited on 09/16/2023).
- Brachthäuser, Jonathan Immanuel, Philipp Schuster, and Klaus Ostermann (Nov. 13, 2020). “Effects as capabilities: effect handlers and lightweight effect polymorphism”. In: *Proceedings of the ACM on Programming Languages* 4 (OOPSLA), pp. 1–30. ISSN: 2475-1421. DOI: 10.1145/3428194. URL: <https://dl.acm.org/doi/10.1145/3428194> (visited on 06/02/2023).
- Brady, Edwin (Sept. 25, 2013). “Programming and reasoning with algebraic effects and dependent types”. In: *Proceedings of the 18th ACM SIGPLAN international conference on Functional programming*. ICFP’13: ACM SIGPLAN International Conference on Functional Programming. Boston Massachusetts USA: ACM, pp. 133–144. ISBN: 978-1-4503-2326-0. DOI: 10.1145/2500365.2500581. URL: <https://dl.acm.org/doi/10.1145/2500365.2500581> (visited on 06/13/2023).
- Dijkstra, Edsger W. (Mar. 1968). “Letters to the editor: go to statement considered harmful”. In: *Communications of the ACM* 11.3, pp. 147–148. ISSN: 0001-0782, 1557-7317. DOI: 10.1145/362929.362947. URL: <https://dl.acm.org/doi/10.1145/362929.362947> (visited on 05/31/2023).
- Hindley, R. (Dec. 1969). “The Principal Type-Scheme of an Object in Combinatory Logic”. In: *Transactions of the American Mathematical Society* 146, p. 29. ISSN: 00029947. DOI: 10.

- 2307/1995158. URL: <https://www.jstor.org/stable/1995158?origin=crossref> (visited on 09/16/2023).
- Hutton, Graham and Erik Meijer (1996). *Monadic parser combinators*. URL: <https://nottingham-repository.worktribe.com/output/1024440>.
- Kammar, Ohad, Sam Lindley, and Nicolas Oury (Sept. 25, 2013). “Handlers in action”. In: *Proceedings of the 18th ACM SIGPLAN international conference on Functional programming*. ICFP’13: ACM SIGPLAN International Conference on Functional Programming. Boston Massachusetts USA: ACM, pp. 145–158. ISBN: 978-1-4503-2326-0. DOI: 10.1145/2500365.2500590. URL: <https://dl.acm.org/doi/10.1145/2500365.2500590> (visited on 01/18/2023).
- Kiselyov, Oleg and Hiromi Ishii (Jan. 28, 2016). “Freer monads, more extensible effects”. In: *ACM SIGPLAN Notices* 50.12, pp. 94–105. ISSN: 0362-1340, 1558-1160. DOI: 10.1145/2887747.2804319. URL: <https://dl.acm.org/doi/10.1145/2887747.2804319> (visited on 06/03/2023).
- Leijen, Daan (June 5, 2014). “Koka: Programming with Row Polymorphic Effect Types”. In: *Electronic Proceedings in Theoretical Computer Science* 153, pp. 100–126. ISSN: 2075-2180. DOI: 10.4204/EPTCS.153.8. URL: <http://arxiv.org/abs/1406.2061v1> (visited on 06/16/2023).
- (Jan. 2017). “Type directed compilation of row-typed algebraic effects”. In: *Proceedings of the 44th ACM SIGPLAN Symposium on Principles of Programming Languages*. POPL ’17: The 44th Annual ACM SIGPLAN Symposium on Principles of Programming Languages. Paris France: ACM, pp. 486–499. ISBN: 978-1-4503-4660-3. DOI: 10.1145/3009837.3009872. URL: <https://dl.acm.org/doi/10.1145/3009837.3009872> (visited on 04/08/2023).
- (Mar. 15, 2023). *The Koka Programming Language*. URL: <https://koka-lang.github.io/koka/doc/book.html> (visited on 06/13/2023).
- Lindley, Sam, Conor McBride, and Craig McLaughlin (Oct. 3, 2017). *Do be do be do*. arXiv: 1611.09259[cs]. URL: <http://arxiv.org/abs/1611.09259> (visited on 04/08/2023).
- Matsakis, Nicholas D. and Felix S. Klock (Oct. 18, 2014). “The rust language”. In: *Proceedings of the 2014 ACM SIGAda annual conference on High integrity language technology*. HILT ’14: High Integrity Language Technology ACM SIGAda Annual Conference. Portland Oregon USA: ACM, pp. 103–104. ISBN: 978-1-4503-3217-0. DOI: 10.1145/2663171.2663188. URL: <https://dl.acm.org/doi/10.1145/2663171.2663188> (visited on 09/16/2023).
- Milner, Robin (Dec. 1978). “A theory of type polymorphism in programming”. In: *Journal of Computer and System Sciences* 17.3, pp. 348–375. ISSN: 00220000. DOI: 10.1016/0022-0000(78)90014-4. URL: <https://linkinghub.elsevier.com/retrieve/pii/S0022000078900144> (visited on 09/16/2023).
- Moggi, Eugenio (1989a). *An Abstract View of Programming Languages*.
- (1989b). “Computational lambda-calculus and monads”. In: *[1989] Proceedings. Fourth Annual Symposium on Logic in Computer Science*. [1989] Proceedings. Fourth Annual Symposium on Logic in Computer Science. Pacific Grove, CA, USA: IEEE Comput. Soc. Press, pp. 14–23. ISBN: 978-0-8186-1954-0. DOI: 10.1109/LICS.1989.39155. URL: <http://ieeexplore.ieee.org/document/39155/> (visited on 04/08/2023).
- Peyton Jones, Simon L. and Philip Wadler (1993). “Imperative functional programming”. In: *Proceedings of the 20th ACM SIGPLAN-SIGACT symposium on Principles of programming languages - POPL ’93*. the 20th ACM SIGPLAN-SIGACT symposium. Charleston, South Carolina, United States: ACM Press, pp. 71–84. ISBN: 978-0-89791-560-1. DOI: 10.1145/158511.158524. URL: <http://portal.acm.org/citation.cfm?doid=158511.158524> (visited on 06/03/2023).
- Piróg, Maciej et al. (July 9, 2018). “Syntax and Semantics for Operations with Scopes”. In: *Proceedings of the 33rd Annual ACM/IEEE Symposium on Logic in Computer Science*. LICS ’18: 33rd Annual ACM/IEEE Symposium on Logic in Computer Science. Oxford

- United Kingdom: ACM, pp. 809–818. ISBN: 978-1-4503-5583-4. DOI: 10.1145/3209108.3209166. URL: <https://dl.acm.org/doi/10.1145/3209108.3209166> (visited on 09/11/2023).
- Plotkin, Gordon and John Power (2001). “Adequacy for Algebraic Effects”. In: *Foundations of Software Science and Computation Structures*. Ed. by Furio Honsell and Marino Miculan. Red. by Gerhard Goos, Juris Hartmanis, and Jan van Leeuwen. Vol. 2030. Series Title: Lecture Notes in Computer Science. Berlin, Heidelberg: Springer Berlin Heidelberg, pp. 1–24. ISBN: 978-3-540-41864-1 978-3-540-45315-4. DOI: 10.1007/3-540-45315-6\_1. URL: [http://link.springer.com/10.1007/3-540-45315-6\\_1](http://link.springer.com/10.1007/3-540-45315-6_1) (visited on 04/08/2023).
- (2003). “Algebraic Operations and Generic Effects”. In: *Applied Categorical Structures* 11.1, pp. 69–94. ISSN: 09272852. DOI: 10.1023/A:1023064908962. URL: <http://link.springer.com/10.1023/A:1023064908962> (visited on 06/03/2023).
- Plotkin, Gordon and Matija Pretnar (2009). “Handlers of Algebraic Effects”. In: *Programming Languages and Systems*. Ed. by Giuseppe Castagna. Vol. 5502. Series Title: Lecture Notes in Computer Science. Berlin, Heidelberg: Springer Berlin Heidelberg, pp. 80–94. ISBN: 978-3-642-00589-3 978-3-642-00590-9. DOI: 10.1007/978-3-642-00590-9\_7. URL: [http://link.springer.com/10.1007/978-3-642-00590-9\\_7](http://link.springer.com/10.1007/978-3-642-00590-9_7) (visited on 04/08/2023).
- Sivaramakrishnan, K. C. et al. (June 19, 2021). “Retrofitting Effect Handlers onto OCaml”. In: *Proceedings of the 42nd ACM SIGPLAN International Conference on Programming Language Design and Implementation*, pp. 206–221. DOI: 10.1145/3453483.3454039. arXiv: 2104.00250[cs]. URL: <http://arxiv.org/abs/2104.00250> (visited on 04/08/2023).
- Swierstra, Wouter (July 2008). “Data types à la carte”. In: *Journal of Functional Programming* 18.4. ISSN: 0956-7968, 1469-7653. DOI: 10.1017/S0956796808006758. URL: [http://www.journals.cambridge.org/abstract\\_S0956796808006758](http://www.journals.cambridge.org/abstract_S0956796808006758) (visited on 06/11/2023).
- Wadler, Philip (1992). “The essence of functional programming”. In: *Proceedings of the 19th ACM SIGPLAN-SIGACT symposium on Principles of programming languages - POPL ’92*. the 19th ACM SIGPLAN-SIGACT symposium. Albuquerque, New Mexico, United States: ACM Press, pp. 1–14. ISBN: 978-0-89791-453-6. DOI: 10.1145/143165.143169. URL: <http://portal.acm.org/citation.cfm?doid=143165.143169> (visited on 08/24/2023).
- (June 1996). “Lazy versus strict”. In: *ACM Computing Surveys* 28.2, pp. 318–320. ISSN: 0360-0300, 1557-7341. DOI: 10.1145/234528.234738. URL: <https://dl.acm.org/doi/10.1145/234528.234738> (visited on 09/16/2023).
- Wu, Nicolas and Tom Schrijvers (2015). “Fusion for Free: Efficient Algebraic Effect Handlers”. In: *Mathematics of Program Construction*. Ed. by Ralf Hinze and Janis Voigtländer. Vol. 9129. Series Title: Lecture Notes in Computer Science. Cham: Springer International Publishing, pp. 302–322. ISBN: 978-3-319-19796-8 978-3-319-19797-5. DOI: 10.1007/978-3-319-19797-5\_15. URL: [https://link.springer.com/10.1007/978-3-319-19797-5\\_15](https://link.springer.com/10.1007/978-3-319-19797-5_15) (visited on 09/11/2023).
- Wu, Nicolas, Tom Schrijvers, and Ralf Hinze (June 10, 2014). *Effect Handlers in Scope*.
- Xie, Ningning et al. (Oct. 31, 2022). “First-class names for effect handlers”. In: *Proceedings of the ACM on Programming Languages* 6 (OOPSLA2), pp. 30–59. ISSN: 2475-1421. DOI: 10.1145/3563289. URL: <https://dl.acm.org/doi/10.1145/3563289> (visited on 06/13/2023).
- Yang, Zhixuan et al. (2022). “Structured Handling of Scoped Effects: Extended Version”. In: Publisher: arXiv Version Number: 1. DOI: 10.48550/ARXIV.2201.10287. URL: <https://arxiv.org/abs/2201.10287> (visited on 06/13/2023).





# Appendix A

---

## Elaine Example Programs

This chapter contains longer Elaine samples with some additional explanation.

### A.1 A naive SAT solver

This program is a naive brute-forcing SAT solver. We first define a `Yield` effect, so we can yield multiple values from the computation. We will use this to find all possible combinations of boolean inputs that satisfy the formula. The `Logic` effect has two operations. The `branch` operation will call the continuation twice; once with `false` and once `true`. With `fail`, we can indicate that a branch has failed. To find all solutions, we just `branch` on all inputs and `yield` when a correct solution has been found and `fail` when the formula is not satisfied. In the listing below, we check for solutions of the equation  $\neg a \wedge b$ .

```
1 use std;
2
3 effect Yield {
4   yield(String) ()
5 }
6
7 effect Logic {
8   branch() Bool
9   fail() a
10 }
11
12 let hYield = handler {
13   return(x) { "" }
14   yield(m) {
15     concat(concat(m, "\n"), resume(()))
16   }
17 };
18
19 let hLogic = handler {
20   return(x) { () }
21   branch() {
22     resume(true);
23     resume(false)
24   }
25   fail() { () }
26 };
```

Elaine

```

27
28 let show_bools = fn(a, b, c) {
29     let a = concat(show_bool(a), ", ");
30     let b = concat(show_bool(b), ", ");
31     concat(concat(a, b), show_bool(c))
32 };
33
34 let f = fn(a, b, c) { and(not(a), b) };
35
36 let assert = fn(f, a, b, c) <Logic,Yield> () {
37     if f(a, b, c) {
38         yield(show_bools(a, b, c))
39     } else {
40         fail()
41     }
42 };
43
44 let main = handle[hYield] handle[hLogic] {
45     assert(f, branch(), branch(), branch());
46 };

```

## A.2 Reader Effect

The implementation of the reader effect is a standard application for higher-order effects. We start with a higher-order `Reader!` effect with an operation `local!` and an algebraic `Ask` effect. The `local!` operation is elaborated into a computation that handles the `Ask` with the modified value.

This effect corresponds to the `Reader` monad as defined by Haskell's `mtl` library.

```

1 use std;
2
3 effect Ask {
4     ask() Int
5 }
6
7 effect Reader! {
8     local!(fn(Int) Int, a) a
9 }
10
11 let hAsk = fn(v: Int) {
12     handler {
13         return(x) { x }
14         ask() { resume(v) }
15     }
16 };
17
18 let eReader = elaboration Reader! -> <Ask> {
19     local!(f, c) {
20         handle[hAsk(f(ask()))] c
21     }
22 };

```

Elaine

```

23
24 let double = fn(x) { mul(2, x) };
25
26 let main = handle[hAsk(2)] elab[eReader] {
27     local!(double, add(ask(), ask()));
28 };

```

### A.3 Writer Effect

The implementation of the writer effect is similar to the implementation of the reader effect. Again, we elaborate a higher-order effect, `Writer!`, into an algebraic effect, `Out`, with a subset of the operations. The higher-order `ensor!` operation handles the algebraic effect to access the output and applies the censoring function to it.

This effect corresponds to the `Writer` monad as defined by Haskell's `mtl` library.

```

1 use std;
2
3 effect Writer! {
4     censor!(fn(String) String, a) a
5     tell!(String) ()
6 }
7
8 effect Out {
9     tell(String) ()
10 }
11
12 type Output[a] {
13     Output(String, a)
14 }
15
16 let hOut = handler {
17     return(x) { Output("", x) }
18     tell(s) {
19         match resume(()) {
20             Output(s', x) => Output(concat(s, s'), x)
21         }
22     }
23 };
24
25 let eWriter = elaboration Writer! -> <Out> {
26     tell!(s) { tell(s) }
27     censor!(f, c) {
28         match handle[hOut] c {
29             Output(s, x) => {
30                 tell(f(s));
31                 x
32             }
33         }
34     }
35 };
36

```

```

37 let main = handle[hOut] elab {
38   tell("foo");
39   censor!(fn(s) { "bar" }, {
40     tell("baz");
41     5
42   });
43 };

```

## A.4 Structured Logging

Since higher-order effects are suitable for delimiting the scope of effects, we can make an effect for structured logging. The idea is that every **log** call appends a message to the output, but the message is prefixed with some context. This context starts out as the empty string, but within every **context!** call, a string is added to this context.

```

1  use std;
2
3  effect Write {
4    write(String) ()
5  }
6
7  effect Read {
8    ask() String
9  }
10
11 effect Log! {
12   context!(String, a) a
13   log!(String) ()
14 }
15
16 let hRead = fn(v: String) {
17   handler {
18     return(x) { x }
19     ask() { resume(v) }
20   }
21 };
22
23 let hWrite = handler {
24   return(x) { "" }
25   write(m) {
26     let rest = resume(());
27     let msg = concat(m, "\n");
28     concat(msg, rest)
29   }
30 };
31
32 let eLog = elaboration Log! -> <Read,Write> {
33   context!(s, c) {
34     let new_context = concat(concat(ask(), s), ":");
35     handle[hRead(new_context)] c
36   }

```

Elaine

```

37     log!(m) {
38         write(concat(concat(ask(), " "), m))
39     }
40 };
41
42 let main = handle[hRead("")] handle[hWrite] elab[eLog] {
43     context!("main", {
44         log!("msg1");
45         context!("foo", {
46             log!("msg2")
47         });
48         context!("bar", {
49             log!("msg3")
50         })
51     })
52 };

```

## A.5 Parser Combinators

Monadic parser combinators (Hutton and Meijer 1996) are a popular technique for constructing parsers. The parser for Elaine is also written using `megaparsec`<sup>1</sup>, which is a monadic parser combinator library for Haskell. Attempts have been made to implement parser combinators using algebraic effects. However, it requires higher-order combinators for a full feature set matching that of monadic parser combinators. For example, the `alt` combinator takes two branches and attempts to parse the first branch and tries the second branch if the first one fails. This is remarkably similar to the `catch` operation of the exception effect and is indeed higher-order.

Below is a full listing of a JSON parser written in Elaine using a variation on parser combinators using effects. It is implemented using a higher-order `Parse!` effect, which is elaborated into a state and an abort effect, which are imported from the standard library. The `try!` effect is a higher-order effect which takes an effectful computation as an argument. It applies the computation and returns its value if it succeeds, otherwise it will reset the state and return `Nothing()`.

Higher-order effects are convenient for parser combinators, but not necessary. Instead of the `try!` operation, the non-determinism effect can be used to write a backtracking parser. An implementation of that technique in Effekt available at <https://effekt-lang.org/docs/casestudies/parser>.

```

1 use std;
2 use maybe;
3 use list;
4 use state_str;
5 use abort;
6
7 effect Parse! {
8     # Signal that this branch has failed to parse
9     fail!() a
10    # Try to apply the parser, reset the state if it fails
11    try!(a) Maybe[a]

```

<sup>1</sup><https://github.com/mrkkp/megaparsec>

```

12     # Remove and return the first character of the input
13     eat!() String
14 }
15
16 let eParse = elaboration Parse! -> <State,Abort> {
17     fail!() { abort() }
18     try!(x) {
19         let old_state = get();
20         match handle[hAbort] x {
21             Just(res) => Just(res),
22             Nothing() => {
23                 put(old_state);
24                 Nothing()
25             }
26         }
27     }
28     eat!() {
29         let state = get();
30         put(drop(1, state));
31         take(1, state)
32     }
33 };
34
35 ### Combinators
36 let alt2 = fn(a, b) {
37     match try!(a()) {
38         Just(x) => x,
39         Nothing() => b(),
40     }
41 };
42
43 let rec alt = fn(parsers) {
44     match parsers {
45         Cons(p, ps) => alt2(p, fn() { alt(ps) }),
46         Nil() => fail!(),
47     }
48 };
49
50 let rec many = fn(p) {
51     match try!(p()) {
52         Just(x) => Cons(x, many(p)),
53         Nothing() => Nil(),
54     }
55 };
56
57 let separated = fn(
58     p: fn() <Parse!> a,
59     separator: fn() <Parse!> b,
60 ) <Parse!> List[a] {
61     match try!(p()) {
62         Just(x) => Cons(x, many(fn() {separator(); p()})),

```

```

63     Nothing() => Nil()
64   }
65 };
66
67 ### Parsers
68 # Parse a token specified as a string
69 let token = fn(s) {
70   let c = eat!();
71   if str_eq(s, c) {
72     c
73   } else {
74     fail!()
75   }
76 };
77
78 let rec contains_str = fn(s, l) {
79   match l {
80     Cons(x, xs) => {
81       if str_eq(x, s) {
82         true
83       } else {
84         contains_str(s, xs)
85       }
86     },
87     Nil() => false,
88   }
89 };
90
91 let one_of = fn(s) {
92   let list_of_chars = explode(s);
93   fn() {
94     let c = eat!();
95     if contains_str(c, list_of_chars) {
96       c
97     } else {
98       fail!()
99     }
100   }
101 };
102
103 # Parse a single digit
104 let digit = one_of("0123456789");
105 let str_char = one_of(join([
106   "0123456789",
107   "ABCDEFGHIJKLMNOPQRSTUVWXYZ",
108   "abcdefghijklmnopqrstuvwxyz",
109   "_- ?!",
110 ]));
111 let white_one = one_of(" \n\t");
112 let white = fn() { many(white_one); () };
113

```



```
114 let tokenws = fn(s) {
115     let t = token(s);
116     white();
117     t
118 };
119
120 let comma_separated = fn(p: fn() <Parse!> a) <Parse!> List[a] {
121     separated(p, fn() { tokenws(",") })
122 };
123
124 # Parse as many digits as possible
125 let number = fn() { join(many(digit)) };
126
127 type Json {
128     JsonString(String),
129     JsonInt(String),
130     JsonArray(List[Json]),
131     JsonObject(List[(String, Json)]),
132 }
133
134 let string = fn() <Parse!> String {
135     token("\");
136     let s = join(many(str_char));
137     tokenws("\");
138     s
139 };
140
141 let key_value = fn(value: fn() <Parse!> Json) <Parse!> (String, Json) {
142     let k = string();
143     tokenws(":");
144     (k, value())
145 };
146
147 let object = fn(value: fn() <Parse!> Json) <Parse!> Json {
148     tokenws("{");
149     let kvs = comma_separated(fn() { key_value(value) });
150     tokenws("}");
151     JsonObject(kvs)
152 };
153
154 let array = fn(value) {
155     tokenws("[");
156     let values = comma_separated(value);
157     tokenws("]");
158     JsonArray(values)
159 };
160
161 let rec value = fn() {
162     alt([
163         fn() { array(value) },
164         fn() { object(value) },
```

```
165     fn() { JsonString(string()) },
166     fn() { JsonInt(number()) },
167   ])
168 };
169
170 let parse = fn(parser, input) {
171   let f = handle[hState] handle[hAbort] elab[eParse] parser();
172   f(input)
173 };
174
175 let main = parse(
176   value,
177   "{\"key1\": 123, \"key2\": [1,2,3], \"key3\": \"some string\"}"
178 );
```